
Wolfgang Lux6

1 Introduction

Curry is a universal programming language aiming at the amalgamation of the most impor-
tant declarative programming paradigms, namely functional programming and logic programming.
Curry combines in a seamless way features from functional programming (nested expressions, lazy
evaluation, higher-order functions), logic programming (logical variables, partial data structures,
built-in search), and concurrent programming (concurrent evaluation of constraints with synchro-
nization on logical variables). Moreover, Curry provides additional features in comparison to the
pure languages (compared to functional programming: search, computing with partial information;
compared to logic programming: more efficient evaluation due to the deterministic evaluation of
functions). Moreover, it also amalgamates the most important operational principles developed
in the area of integrated functional logic languages: “residuation” and “narrowing” (see [22] for a
survey on functional logic programming).

The development of Curry is an international initiative intended to provide a common platform
for the research, teaching1 and application of integrated functional logic languages. This document
describes the features of Curry, its syntax and operational semantics.

1Actually, Curry has been successfully applied to teach functional and logic programming techniques in a single

course without switching between different programming languages. More details about this aspect can be found in

[23].

2 Programs

A Curry program specifies the semantics of expressions, where goals, which occur in logic pro-
gramming, are particular expressions (of type Success, cf. Section 4.1.2, page 17). Executing a
Curry program means simplifying an expression until a value (or solution) is computed. To dis-
tinguish between values and reducible expressions, Curry has a strict distinction between (data)
constructors and operations or defined functions on these data. Hence, a Curry program consists
of a set of type and function declarations. The type declarations define the computational domains
(constructors) and the function declarations the operations on these domains. Predicates in the
logic programming sense can be considered as constraints, i.e., functions with result type Success.

Modern functional languages (e.g., Haskell [41], SML [36]) allow the detection of many pro-
gramming errors at compile time by the use of polymorphic type systems. Similar type systems
are also used in modern logic languages (e.g., Gödel [27], λProlog [38]). Curry follows the same
approach, i.e., it is a strongly typed language with a Hindley/Milner-like polymorphic type system
[14].2 Each object in a program has a unique type, where the types of variables and operations can
be omitted and are reconstructed by a type inference mechanism.

2.1 Datatype Declarations

A datatype declaration has the form

data T α1 ... αn = C1 τ11 ... τ1n1 | · · · | Ck τk1 ... τknk

and introduces a new n-ary type constructor T and k new (data) constructors C1, . . . , Ck, where
each Ci has the type

τi1 -> · · · -> τini -> T α1 . . . αn

(i = 1, . . . , k). Each τij is a type expression built from the type variables α1, . . . , αn and some
type constructors. Curry has a number of built-in type constructors, like Bool, Int, -> (function
space), or, lists and tuples, which are described in Section 4.1. Since Curry is a higher-order
language, the types of functions (i.e., constructors or operations) are written in their curried form
τ1 -> τ2 -> · · · -> τn -> τ where τ is not a functional type. In this case, n is called the arity of
the function. For instance, the datatype declarations

data Bool = True | False

data List a = [] | a : List a

data Tree a = Leaf a | Node (Tree a) a (Tree a)

introduces the datatype Bool with the 0-ary constructors (constants) True and False, and the
polymorphic types List a and Tree a of lists and binary trees. Here, “:” is an infix operator, i.e.,
“a:List a” is another notation for “(:) a (List a)”. Lists are predefined in Curry, where the
notation “[a]” is used to denote list types (instead of “List a”). The usual convenient notations
for lists are supported, i.e., [0,1,2] is an abbreviation for 0:(1:(2:[])) (see also Section 4.1).

A data term is a variable x or a constructor application c t1 . . . tn where c is an n-ary constructor
and t1, . . . , tn are data terms. An expression is a variable or a (partial) application ϕ e1 . . . em

2The extension of this type system to Haskell’s type classes is not included in the kernel language but could be

considered in a future version.

where ϕ is a function or constructor and e1, . . . , em are expressions. A data term or expression is
called ground if it does not contain any variable. Ground data terms correspond to values in the
intended domain, and expressions containing operations should be evaluated to data terms. Note
that traditional functional languages compute on ground expressions, whereas logic languages also
allow non-ground expressions.

2.2 Type Synonym Declarations

To make type definitions more readable, it is possible to specify new names for type expressions by
a type synonym declaration. Such a declaration has the general form

type T α1 ... αn = τ

which introduces a new n-ary type constructor T . α1, . . . , αn are pairwise distinct type variables
and τ is a type expressions built from type constructors and the type variables α1, . . . , αn. The type
(T τ1 . . . τn) is equivalent to the type {α1 7→ τ1, . . . , αn 7→ τn}(τ), i.e., the type expression τ where
each αi is replaced by τi. Thus, a type synonym and its definition are always interchangeable and
have no influence on the typing of a program. For example, we can provide an alternative notation
for list types and strings by the following type synonym declarations:

type List a = [a]

type Name = [Char]

Since a type synonym introduces just another name for a type expression, recursive or mutually
dependent type synonym declarations are not allowed. Therefore, the following declarations are
invalid :

type RecF a = a -> RecF a -- recursive definitions not allowed

type Place = [Addr] -- mutually recursive definitions not allowed

type Addr = [Place]

However, recursive definitions with an intervening datatype are allowed, since recursive datatype
definitions are also allowed. For instance, the following definitions are valid:

type Place = [Addr]

data Addr = Tag [Place]

2.3 Function Declarations

A function is defined by a type declaration (which can be omitted) followed by a list of defining
equations. A type declaration for an n-ary function has the form

f :: τ1 -> τ2 -> · · · -> τn -> τ

where τ1, . . . , τn, τ are type expressions and τ is not a functional type. The simplest form of a
defining equation (or rule) for an n-ary function f (n ≥ 0) is

f t1 . . . tn = e

5



where t1, . . . , tn are data terms (also called patterns) and the right-hand side e is an expression.
The left-hand side f t1 . . . tn must not contain multiple occurrences of a variable. Functions can
also be defined by conditional equations which have the form

f t1 . . . tn | c = e

where the condition c is a constraint (cf. Section 2.6). In order to apply a conditional equation, its
condition must be solved.

Note that one can define functions with a non-determinate behavior by providing several rules
with overlapping left-hand sides or free variables (i.e., variables which do not occur in the left-hand
side) in the conditions or right-hand sides of rules. For instance, the following non-deterministic
function inserts an element at an arbitrary position in a list:

insert x [] = [x]

insert x (y:ys) = x : y : ys

insert x (y:ys) = y : insert x ys

Such non-deterministic functions can be given a perfect declarative semantics [19] and their imple-
mentation causes no overhead in Curry since techniques for handling non-determinism are already
contained in the logic programming part (see also [5]). However, deterministic functions are advan-
tageous since they provide for more efficient implementations (like the dynamic cut [34]). If one is
interested only in defining deterministic functions, this can be ensured by the following restrictions:

1. Each variable occurring in the right-hand side of a rule must also occur in the corresponding
left-hand side.

2. If f t1 . . . tn | c = e and f t′1 . . . t′n | c′ = e′ are rules defining f and σ is a substitu-
tion3 with σ(t1 . . . tn) = σ(t′1 . . . t′n), then at least one of the following conditions holds:

(a) σ(e) = σ(e′) (compatibility of right-hand sides).

(b) σ(c) and σ(c′) are not simultaneously satisfiable (incompatibility of conditions). A de-
cidable approximation of this condition can be found in [30].

These conditions ensure the confluence of the rules if they are considered as a conditional term
rewriting system [47]. Implementations of Curry may check these conditions and warn the user if
they are not satisfied. There are also more general conditions to ensure confluence [47] which can
be checked instead of the above conditions.

Note that defining equations of higher-type, e.g., f = g if f and g are of type Bool -> Bool, are
formally excluded in order to define a simple operational semantics based on first-order rewriting.4

For convenience, a defining equation f = g between functions is allowed but will be interpreted in
Curry as syntactic sugar for the corresponding defining equation f x = g x on base types.

3A substitution σ is a mapping from variables into expressions which is extended to a homomorphism on expressions

by defining σ(f t1 . . . tn) = f σ(t1) . . . σ(tn). {x1 7→ e1, . . . , xk 7→ ek} denotes the substitution σ with σ(xi) = ei

(i = 1, . . . , k) and σ(x) = x for all variables x 6∈ {x1, . . . , xk}.
4Note that this is also necessary in an extension of Curry which allows higher-order rewrite rules, since rules with

lambda-abstractions in left-hand sides which are not of base type may cause a gap between the standard notion of

higher-order rewriting and the corresponding equational theory [40].

2.3.1 Functions vs. Variables

In lazy functional languages, different occurrences of the same variable are shared to avoid multiple
evaluations of identical expressions. For instance, if we apply the rule

double x = x+x

to an expression double t, we obtain the new expression t+t but both occurrences of t denote the
identical expression, i.e., these subterms will be simultaneously evaluated. Thus, several occurrences
of the same variable are always shared, i.e., if one occurrence of an argument variable, which
might bound to an evaluable expression when the function is applied, is evaluated to some value,
all other occurrences of this variable are replaced by the same value (without evaluating these
other occurrences again). This sharing is necessary not only for efficiency reasons but it has also
an influence on the soundness of the operational semantics in the presence of non-deterministic
functions (see also [19]). For instance, consider the non-deterministic function coin defined by the
rules

coin = 0

coin = 1

Thus, the expression coin evaluates to 0 or 1. However, the result values of the expression
(double coin) depend on the sharing of the two occurrences of coin after applying the rule
for double: if both occurrences are shared (as in Curry), the results are 0 or 2, otherwise (without
sharing) the results are 0, 1, or 2. The sharing of argument variables corresponds to the so-called
“call time choice” in the declarative semantics [19]: if a rule is applied to a function call, a unique
value must be assigned to each argument (in the example above: either 0 or 1 must be assigned
to the occurrence of coin when the expression (double coin) is evaluated). This does not mean
that functions have to be evaluated in a strict manner but this behavior can be easily obtained by
sharing the different occurrences of a variable.

Since different occurrences of the same variable are always shared but different occurrences of
(textually identical) function calls are not shared, it is important to distinguish between variables
and functions. Usually, all symbols occurring at the top level in the left-hand side of some rule are
considered as functions and the non-constructor symbols in the arguments of the left-hand sides
are considered as variables. But note that there is a small exception from this general rule in local
declarations (see Section 2.4).

2.3.2 Conditional Equations with Boolean Guards

For convenience and compatibility with Haskell, one can also write conditional equations with
multiple guards

f t1 . . . tn | b1 = e1
...

| bk = ek

where b1, . . . , bk (k > 0) are expressions of type Bool. Such a rule is interpreted as in Haskell:
the guards are successively evaluated and the right-hand side of the first guard which is evaluated
to True is the result of applying this equation. Thus, this equation can be considered as an

7



abbreviation for the rule

f t1 . . . tn = if b1 then e1 else
...

if bk then ek else failed

(where failed is a non-reducible function defined in the prelude). To write rules with several
Boolean guards more nicely, there is a Boolean function otherwise which is predefined as True.
For instance, the factorial function can be declared as follows:

fac n | n==0 = 1

| otherwise = fac(n-1)*n

Since all guards have type Bool, this definition is equivalent to (after a simple optimization)

fac n = if n==0 then 1 else fac(n-1)*n

To avoid confusion, it is not allowed to mix the Haskell-like notation with Boolean guards and the
standard notation with constraints: in a rule with multiple guards, all guards must be expressions
of type Bool. The default type of a guard in single-guarded rules is Success, i.e., in a rule like

f c x | c = x

the type of the variable c is Success (provided that it is not restricted to Bool by the use of f).

2.3.3 As-Patterns

The patterns in a defining equation, i.e., the arguments of the left-hand sides, are required to be
data terms without multiple occurrences of variables. Patterns are useful to split the definition of
functions into easily comprehensible cases. Thus, a pattern denotes some part of a structure of
the actual argument. Sometimes one wants to reuse this structure in the right-hand side of the
defining equation. This can be expressed by an as-pattern which allows to identify this structure by
a variable. An as-pattern has the form v@pat where the variable v identifies the structure matched
by the pattern pat. For instance,

dropFalse (False:ys) = ys

dropFalse xs@(True:_) = xs

is equivalent to

dropFalse0 xs = dropFalse’ xs

where

dropFalse’ (False:ys) = ys

dropFalse’ (True:_) = xs

(local declarations introduced by where-clauses are discussed in Section 2.4). However, as-patterns
are usually implemented more efficiently without introducing an auxiliary function.

2.4 Local Declarations

Since not all auxiliary functions should be globally visible, it is possible to restrict the scope of
declared entities. Note that the scope of parameters in function definitions is already restricted
since the variables occurring in parameters of the left-hand side are only visible in the corresponding
conditions and right-hand sides. The visibility of other entities can be restricted using let in
expressions or where in defining equations.

An expression of the form let decls in exp introduces a set of local names. The list of local
declarations decls can contain function definitions as well as definitions of constants by pattern
matching. The names introduced in these declarations are visible in the expression exp and the
right-hand sides of the declarations in decls, i.e., the local declarations can be mutually recursive.
For instance, the expression

let a=3*b

b=6

in 4*a

reduces to the value 72.
Auxiliary functions which are only introduced to define another function should often not be

visible outside. Therefore, such functions can be declared in a where-clause added to the right-
hand side of the corresponding function definition. This is done in the following definition of a fast
exponentiation where the auxiliary functions even and square are only visible in the right-hand
side of the rule for exp:

exp b n = if n==0

then 1

else if even n then square (exp b (n ‘div‘ 2))

else b * (exp b (n-1))

where even n = n ‘mod‘ 2 == 0

square n = n*n

Similarly to let, where-clauses can contain mutually recursive function definitions as well as defi-
nitions of constants by pattern matching. The names declared in the where-clauses are only visible
in the corresponding conditions and right-hand sides. As a further example, the following Curry
program implements the quicksort algorithm with a function split which splits a list into two lists
containing the smaller and larger elements:

split e [] = ([],[])

split e (x:xs) | e>=x = (x:l,r)

| e<x = (l,x:r)

where

(l,r) = split e xs

qsort [] = []

qsort (x:xs) = qsort l ++ (x:qsort r)

where

(l,r) = split x xs

9



To distinguish between locally introduced functions and variables (see also Section 2.3.1), we define
a local pattern as a (variable) identifier or an application where the top symbol is a data constructor.
If the left-hand side of a local declaration in a let or where is a pattern, then all identifiers in
this pattern that are not data constructors are considered as variables. For instance, the locally
introduced identifiers a, b, l, and r in the previous examples are variables whereas the identifiers
even and square denote functions. Note that this rule exclude the definition of 0-ary local functions
since a definition of the form “where f = ...” is considered as the definition of a local variable
f by this rule which is usually the intended interpretation (see previous examples). Appendix D.7
contains a precise formalization of the meaning of local definitions.

2.5 Free Variables

Since Curry is intended to cover functional as well as logic programming paradigms, expressions
(or constraints, see Section 2.6) might contain free (unbound, uninstantiated) variables. The idea
is to compute values for these variables such that the expression is reducible to a data term or that
the constraint is solvable. For instance, consider the definitions

mother John = Christine

mother Alice = Christine

mother Andrew = Alice

Then we can compute a child of Alice by solving the equation (see Section 2.6)
“mother x =:= Alice”. Here, x is a free variable which is instantiated to Andrew in order to
reduce the equation’s left-hand side to Alice. Similarly, we can compute a grandchild of Chistine
by solving the equation “mother (mother x) =:= Christine” which yields the value Andrew for
x.

In logic programming languages like Prolog, all free variables are considered as existentially
quantified at the top level. Thus, they are always implicitly declared at the top level. In a language
with different nested scopes like Curry, it is not clear to which scope an undeclared variable belongs
(the exact scope of a variable becomes particularly important in the presence of search operators,
see Section 8, where existential quantifiers and lambda abstractions are often mixed). Therefore,
Curry requires that each free variable x must be explicitly declared using a local declaration of the
form x free. The variable is then introduced as unbound with the same scoping rules as for all
other local entities (see Section 2.4). For instance, we can define

isgrandmother g | let c free in mother (mother c) =:= g = True

As a further example, consider the definition of the concatentation of two lists:

append [] ys = ys

append (x:xs) ys = x : append xs ys

Then we can define the function last which computes the last element of a list by the rule

last l | append xs [x] =:= l = x where x,xs free

Since the variable xs occurs in the condition but not in the right-hand side, the following definition
is also possible:

10



last l | let xs free in append xs [x] =:= l = x where x free

Note that the free declarations can be freely mixed with other local declarations after a let or
where. The only difference is that a declaration like “let x free” introduces an unbound variable
whereas other let declarations introduce local functions or parameters. Since all local declarations
can be mutually recursive, it is also possible to use local variables in the bodies of the local functions
in one let declarations. For instance, the following expression is valid (where the functions h and
k are defined elsewhere):

let f x = h x y

y free

g z = k y z

in c y (f (g 1))

Similarly to the usual interpretation of local definitions by lambda lifting [29], this expression can
be interpreted by transforming the local definitions for f and g into global ones by adding the
non-local variables of the bodies as parameters:

f y x = h x y

g y z = k y z

...

let y free in c y (f y (g y 1))

See Appendix D.7 for more details about the meaning and transformation of local definitions.

2.6 Constraints and Equality

A condition of a rule is a constraint which must be solved in order to apply the rule. An elementary
constraint is an equational constraint e1=:=e2 between two expressions (of base type). e1=:=e2 is
satisfied if both sides are reducible to a same ground data term. This notion of equality, which is
the only sensible notion of equality in the presence of non-terminating functions [18, 37] and also
used in functional languages, is called strict equality. As a consequence, if one side is undefined
(nonterminating), then the strict equality does not hold. Operationally, an equational constraint
e1=:=e2 is solved by evaluating e1 and e2 to unifiable data terms. The equational constraint could
also be solved in an incremental way by an interleaved lazy evaluation of the expressions and
binding of variables to constructor terms (see [33] or Section D.3 in the appendix).

Equational constraints should be distinguished from standard Boolean functions (cf. Section 4.1)
since constraints are checked for satisfiability. For instance, the equational constraint [x]=:=[0]

is satisfiable if the variable x is bound to 0. However, the evaluation of [x]=:=[0] does not
deliver a Boolean value True or False, since the latter value would require a binding of x to all
values different from 0 (which could be expressed if a richer constraint system than substitutions,
e.g., disequality constraints [8], is used). This is sufficient since, similarly to logic programming,
constraints are only activated in conditions of equations which must be checked for satisfiability.

Note that the basic kernel of Curry only provides strict equations e1=:=e2 between expressions
as elementary constraints. Since it is conceptually fairly easy to add other constraint structures [35],
extensions of Curry may provide richer constraint systems to support constraint logic programming
applications.

Constraints can be combined into a conjunction written as c1 & c2. The conjunction is in-
terpreted concurrently : if the combined constraint c1 & c2 should be solved, c1 and c2 are solved
concurrently. In particular, if the evaluation of c1 suspends, the evaluation of c2 can proceed which
may cause the reactivation of c1 at some later time (due to the binding of common variables). In
a sequential implementation, the evaluation of c1 & c2 could be started by an attempt to solve c1.
If the evaluation of c1 suspends, an evaluation step is applied to c2.

It is interesting to note that parallel functional computation models [12, 13] are covered by the
use of concurrent constraints. For instance, a constraint of the form

x =:= f t1 & y =:= g t2 & z =:= h x y

specifies a potentially concurrent computation of the functions f, g and h where the function h

can proceed its computation only if the arguments have been bound by evaluating the expressions
f t1 and g t2. Since constraints can be passed as arguments or results of functions (like any other
data object or function), it is possible to specify general operators to create flexible communica-
tion architectures similarly to Goffin [13]. Thus, the same abstraction facilities could be used for
sequential as well as concurrent programming. On the other hand, the clear separation between
sequential and concurrent computations (e.g., a program without any occurrences of concurrent
conjunctions is purely sequential) supports the use of efficient and optimal evaluation strategies for
the sequential parts [5, 7], where similar techniques for the concurrent parts are not available.

2.7 Higher-order Features

Curry is a higher-order language supporting common functional programming techniques by partial
function applications and lambda abstractions. Function application is denoted by juxtaposition
the function and its argument. For instance, the well-known map function is defined in Curry by

map :: (a -> b) -> [a] -> [b]

map f [] = []

map f (x:xs) = f x : map f xs

However, there is an important difference w.r.t. to functional programming. Since Curry is also
a logic language, it allows logical variables also for functional values, i.e., it is possible to evaluate
the equation map f [1 2] =:= [2 3] which has, for instance, a solution {f=inc} if inc is the
increment function on natural numbers. In principle, such solutions can be computed by extending
(first-order) unification to higher-order unification [25, 38, 43]. Since higher-order unification is a
computationally expensive operation, Curry delays the application of unknown functions until the
function becomes known [2, 46].5

In situations where a function is only used a single time, it is tedious to assign a name to it.
For such cases, anonymous functions (λ-abstractions), denoted by

\x1 . . . xn->e

5Note that an unbound functional variable can never be instantiated if all program rules are constructor-based and

the equational constraint =:= denotes equality between data terms. However, extensions of Curry might overcome

this weakness by instantiating unbound functional variables to (type-conform) functions occurring in the program in

order to evaluate an application (as in [20]), or by considering partial applications (i.e., functions calls with less than

the required number of arguments) as data terms (as in [49]).

are provided.

3 Operational Semantics

Curry’s operational semantics is based on the lazy evaluation of expressions combined with a pos-
sible instantiation of free variables occurring in the expression. If the expression is ground, the
operational model is similar to lazy functional languages, otherwise (possibly non-deterministic)
variable instantiations are performed as in logic programming. If an expression contains free vari-
ables, it may be reduced to different values by binding the free variables to different expressions.
In functional programming, one is interested in the computed value, whereas logic programming
emphasizes the different bindings (answers). Thus, we define for the integrated functional logic
language Curry an answer expression as a pair “σ e” consisting of a substitution σ (the answer
computed so far) and an expression e. An answer expression σ e is solved if e is a data term.
Usually, the identity substitution in answer expressions is omitted, i.e., we write e instead of {} e

if it is clear from the context.
Since more than one answer may exist for expressions containing free variables, in general, initial

expressions are reduced to disjunctions of answer expressions. Thus, a disjunctive expression is a
(multi-)set of answer expressions {σ1 e1, . . . , σn en}. For the sake of readability, we write concrete
disjunctive expressions like

{{x = 0, y = 2} 2 , {x = 1, y = 2} 3}

in the form {x=0,y=2} 2 | {x=1,y=2} 3. Thus, substitutions are represented by lists of equations
enclosed in curly brackets, and disjunctions are separated by vertical bars.

A single computation step performs a reduction in exactly one unsolved expression of a dis-
junction (e.g., in the leftmost unsolved answer expression in Prolog-like implementations). If the
computation step is deterministic, the expression is reduced to a new one. If the computation step
is non-deterministic, the expression is replaced by a disjunction of new expressions. The precise
behavior depends on the function calls occurring in the expression. For instance, consider the
following rules:

f 0 = 2

f 1 = 3

The result of evaluating the expression f 1 is 3, whereas the expression f x should be evaluated
to the disjunctive expression

{x=0} 2 | {x=1} 3 .

To avoid superfluous computation steps and to apply programming techniques of modern functional
languages, nested expressions are evaluated lazily, i.e., the leftmost outermost function call is
primarily selected in a computation step. Due to the presence of free variables in expressions, an
expression may have a free variable at a position where a value is demanded. A value is demanded
in an argument of a function call if the left-hand side of some rule has a constructor at this position,
i.e., in order to apply the rule, the actual value at this position must be the constructor. A value
is also demanded in case expressions (see Section 5.3) or in arguments of external functions (see
Section 9), in particular, in the primitive operation ensureNotFree (see Section 5.4). In this
situation there are two possibilities to proceed:

1. Delay the evaluation of this expression until the corresponding free variable is bound (this

corresponds to the residuation principle which is the basis of languages like Escher [31, 32],
Le Fun [2], Life [1], NUE-Prolog [39], or Oz [46]).

2. If the value is demanded by the left-hand sides of some defined function, (non-
deterministically) instantiate the free variable to the different values required by these left-
hand sides and apply reduction steps using the different rules (this corresponds to narrowing
principle which is the basis of languages like ALF [21], Babel [37], K-Leaf [18], LPG [9], or
SLOG [17]).

Since Curry is an attempt to provide a common platform for different declarative programming
styles and both are reasonable in different contexts, Curry supports both alternatives. If the value
of a free variable is demanded by the left-hand sides of a defined function, the second alternative
(instantiation) is used, otherwise the evaluation is suspended and the evaluation proceeds with
another subexpression (usually, another constraint). Free variables can also be instantiated by
constraint solving, e.g., by evaluating an equational constraint e1=:=e2 (see Section 2.6). Most
external functions, i.e., functions that are not defined by explicit program rules but implemented
by primitive code (see Section 9), like arithmetic operators (see Section 4.1.4), suspend if some
argument is an unbound variable.

For instance, consider the function f as defined above. The expression f x is evaluated by
instantiating x to 0 or 1 and applying a reduction step in both cases. This yields the disjunctive
expression

{x=0} 2 | {x=1} 3 .

On the other hand, primitive arithmetic functions like “+” or “*” suspend if some argument is a
free variable, e.g., “2+x” can not be evaluated to some result. However, the use of such functions
is reasonable if there is a “generator” for values for x. This is the case in the following constraint:

2+x =:= y & f x =:= y

Here, the first constraint 2+x =:= y cannot be evaluated and suspends, but the second constraint
f x =:= y can be solved by binding x to 1, evaluating f 1 to 3, and binding y to 3. After this
binding (or even after binding x), the first constraint can be evaluated and the entire constraint is
solved. Thus, the constraint is solved by the following steps (note that this is not the only possible
order of steps):

2+x =:= y & f x =:= y

; {x=1} 2+1 =:= y & 3 =:= y

; {x=1,y=3} 2+1 =:= 3

; {x=1,y=3} 3 =:= 3

; {x=1,y=3}

(The empty constraint is omitted in the final answer.)

4 Types

4.1 Built-in Types

This section describes the types that are predefined in Curry. For each type, some important
operations are discussed. The complete definition of all operations can be found in the standard
prelude (see Appendix B).

4.1.1 Boolean Values

Boolean values are predefined by the datatype declaration

data Bool = True | False

The (sequential) conjunction is predefined as the left-associative infix operator &&:

(&&) :: Bool -> Bool -> Bool

True && x = x

False && x = False

Similarly, the (sequential) disjunction || and the negation not are defined as usual (see Ap-
pendix B). Furthermore, the function otherwise is predefined as True to write rules with multiple
guards more nicely.

Boolean values are mainly used in conditionals, i.e., the conditional function if_then_else is
predefined as

if_then_else :: Bool -> a -> a -> a

if_then_else b t f = case b of True -> t

False -> f

where “if b then x else y” is syntactic sugar for the application (if_then_else b x y). Due
to the definition based on a case expression (see Section 5.3), the evaluation of a conditional
suspends of the discriminating Boolean argument is unknown.

A function with result type Bool is often called a predicate (in contrast to constraints which
have result type Success, see below).6 There are a number of built-in predicates for comparing
objects, like the predicate “<” to compare numbers (e.g., 1<2 evaluates to True and 4<3 evaluates
to False). There is also a standard predicate “==” to test the convertibility of expressions to
identical data terms (“strict equality”, cf. Section 2.6): e1==e2 reduces to True if e1 and e2 are
reducible to identical ground data terms, and it reduces to False if e1 and e2 are reducible to
different ground data terms. The evaluation of e1==e2 suspends if one of the arguments is a free
variable. If neither e1 nor e2 is a free variable, e1==e2 is reduced according to the following rules:

C == C = True % for all 0-ary constructors C

C x1 . . . xn == C y1 . . . yn = x1==y1 &&...&& xn==yn % for all n-ary constructors C

C x1 . . . xn == D y1 . . . ym = False % for all different constructors C and D

This implements a test for strict equality (cf. Section 2.6). For instance, the test “[0]==[0]”
6Predicates in the logic programming sense should be considered as constraints since they are only checked for

satisfiability and usually not reduced to True or False in contrast to Boolean functions.

reduces to True, whereas the test “1==0” reduces to False. An equality test with a free variable in
one side is delayed in order to avoid an infinite set of solutions for insufficiently instantiated tests
like x==y. Usually, strict equality is only defined on data terms, i.e., “==” is not really polymorphic
but an overloaded function symbol. This could be more precisely expressed using type classes which
might be considered in a future extensions of Curry.

Note that e1==e2 only tests the identity of e1 and e2 but never binds one side to the other if it
is a free variable. This is in contrast to solving the equational constraint e1=:=e2 which is checked
for satisfiability and propagates new variable bindings in order to solve this constraint. Thus, in
terms of concurrent constraint programming languages [44], == and =:= correspond to ask and tell
equality constraints, respectively.

4.1.2 Constraints

The type Success is the result type of expressions used in conditions of defining rules. Since
conditions must be successfully evaluated in order to apply the rule, the type Success can also be
interpreted as the type of successful evaluations. A function with result type Success is also called
a constraint .

Constraints are different from Boolean-valued functions: a Boolean expression reduces to True

or False whereas a constraint is checked for satisfiability. A constraint is applied (i.e., solved) in a
condition of a conditional equation. The equational constraint e1=:=e2 is an elementary constraint
which is solvable if the expressions e1 and e2 are evaluable to unifiable data terms. success denotes
an always solvable constraint.

Constraints can be combined into a conjunction of the form c1 & c2 & . . . & cn by applying the
concurrent conjunction operator &. In this case, all constraints in the conjunction are evaluated
concurrently. Constraints can also be attached to arbitrary expressions by the operator &>, i.e.,
the constrained expression c &> e is evaluated by first solving constraint c and then evaluating
e. This operator can also be used to enforce a sequential order for constraint evaluation, e.g., the
combined constraint c1 &> c2 will be evaluated by first completely evaluating c1 and then c2.

Constraints can be passed as parameters or results of functions like any other data object. For
instance, the following function takes a list of constraints as input and produces a single constraint,
the concurrent conjunction of all constraints of the input list:

conj :: [Success] -> Success

conj [] = success

conj (c:cs) = c & conj cs

The trivial constraint success is usually not shown in answers to a constraint expression. For
instance, the constraint x*x=:=y & x=:=2 is evaluated to the answer

{x=2, y=4}

4.1.3 Functions

The type t1 -> t2 is the type of a function which produces a value of type t2 for each argument
of type t1. A function f is applied to an argument x by writing “f x”. The type expression

t1 -> t2 -> · · · -> tn+1

is an abbreviation for the type

t1 -> (t2 -> (· · · -> tn+1))

and denotes the type of a (currified) n-ary function, i.e., -> associates to the right. Similarly, the
expression

f e1 e2 . . . en

is an abbreviation for the expression

(...((f e1) e2) . . . en)

and denotes the application of a function f to n arguments, i.e., the application associates to the
left.

The prelude also defines a right-associative application operator “$” which is sometimes useful to
avoid brackets. Since $ has low, right-associative binding precedence, the expression “f $ g $ 3+4”
is equivalent to “f (g (3+4))”.

Furthermore, the prelude also defines right-associative application operators that enforces the
evaluation of the argument to a particular degree. For instance, the definition of “$!” is based
on a predefined (infix) operator seq that evaluates the first argument and returns the value of the
second argument:

($!) :: (a -> b) -> a -> b

f $! x = x ‘seq‘ f x

Thus, if inf is a non-terminating function (e.g., defined by “inf = inf”) and f a constant function
defined by “f _ = 0”, then the evaluation of the expression “f $! inf” does not terminate whereas
the expression “f $ inf” evaluates to 0. Similarly, the operator “$!!” completely evaluates its
second argument (i.e., to a normal form), “$#” evaluates its second argument to a non-variable term
(by ensureNotFree, see Section 5.4) and suspends, if necessary, and “$##” completely evaluates
its second argument to a term without variables.

4.1.4 Integers

The common integer values, like “42” or “-15”, are considered as constructors (constants) of type
Int. The usual operators on integers, like + or *, are predefined functions that are evaluated only
if both arguments are integer values, otherwise such function calls are suspended. Thus, these
functions can be used as “passive constraints” which become active after binding their arguments.
For instance, if the constraint digit is defined by the equations

digit 0 = success

...

digit 9 = success

then the constraint x*x=:=y & x+x=:=y & digit x is solved by delaying the two equations
which will be activated after binding the variable x to a digit by the constraint digit x. Thus, the
corresponding computed solution is

{x=0,y=0} | {x=2,y=4}

4.1.5 Floating Point Numbers

Similarly to integers, values like “3.14159” or “5.0e-4” are considered as constructors of type Float.
Since overloading is not included in the kernel version of Curry, the names of arithmetic functions
on floats are different from the corresponding functions on integers.

4.1.6 Lists

The type [t] denotes all lists whose elements are values of type t. The type of lists can be
considered as predefined by the declaration

data [a] = [] | a : [a]

where [] denotes the empty list and x:xs is the non-empty list consisting of the first element x

and the remaining list xs. Since it is common to denote lists with square brackets, the following
convenient notation is supported:

[e1,e2,...,en]

denotes the list e1:e2:· · ·:en:[] (which is equivalent to e1:(e2:(· · ·:(en:[])...)) since “:” is a
right-associative infix operator). Note that there is an overloading in the notation [t]: if t is a
type, [t] denotes the type of lists containing elements of type t, where [t] denotes a single element
list (with element t) if t is an expression. Since there is a strong distinction between occurrences
of types and expressions, this overloading can always be resolved.

For instance, the following predefined functions define the concatenation of two lists and the
application of a function to all elements in a list:

(++) :: [a] -> [a] -> [a]

[] ++ ys = ys

(x:xs) ++ ys = x : xs ++ ys

map :: (a -> b) -> [a] -> [b]

map f [] = []

map f (x:xs) = f x : map f xs

4.1.7 Characters

Values like ’a’ or ’0’ denote constants of type Char. Special characters can be denoted with a
leading backslash, e.g., ’\n’ for the character with ASCII value 10, or ’\228’ for the character
’ä’ with ASCII value 228.

There are two conversion functions between characters and their corresponding ASCII values:

ord :: Char -> Int

chr :: Int -> Char

4.1.8 Strings

The type String is an abbreviation for [Char], i.e., strings are considered as lists of characters.
String constants are enclosed in double quotes. Thus, the string constant "Hello" is identical to
the character list [’H’,’e’,’l’,’l’,’o’]. A term can be converted into a string by the function

show :: a -> String

For instance, the result of (show 42) is the character list [’4’,’2’].

4.1.9 Tuples

If t1, t2, . . . , tn are types and n ≥ 2, then (t1,t2,...,tn) denotes the type of all n-tuples. The
elements of type (t1,t2,...,tn) are (x1,x2,...,xn) where xi is an element of type ti (i = 1, . . . , n).
Thus, for each n, the tuple-building operation (,. . .,) (with n − 1 commas) can be considered as
an n-ary constructor introduced by the pseudo-declaration

data (a1,a2, . . . ,an) = (,. . .,) a1 a2...an

where (x1,x2,...,xn) is equivalent to the constructor application “(,. . .,) x1 x2...xn”.
The unit type () has only a single element () and can be considered as defined by

data () = ()

Thus, the unit type can also be interpreted as the type of 0-tuples.

4.2 Type System

Curry is a strongly typed language with a Hindley/Milner-like polymorphic type system [14].7 Each
variable, constructor and operation has a unique type, where only the types of constructors have
to be declared by datatype declarations (see Section 2.1). The types of functions can be declared
(see Section 2.3) but can also be omitted. In the latter case they will be reconstructed by a type
inference mechanism.

Note that Curry is an explicitly typed language, i.e., each function has a type. The type can only
be omitted if the type inferencer is able to reconstruct it and to insert the missing type declaration.
In particular, the type inferencer can reconstruct only those types which are visible in the module
(cf. Section 6). Each type inferencer of a Curry implementation must be able to insert the types of
the parameters and the free variables (cf. Section 2.5) for each rule. The automatic inference of the
types of the defined functions might require further restrictions depending on the type inference
method. Therefore, the following definition of a well-typed Curry program assumes that the types
of all defined functions are given (either by the programmer or by the type inferencer). A Curry
implementation must accept a well-typed program if all types are explicitly provided but should
also support the inference of function types according to [14].

A type expression is either a type variable, a basic type like Bool, Success, Int, Float, Char
(or any other type constructor of arity 0), or a type constructor application of the form (T τ1 . . . τn)

7The extension of the type system to Haskell’s type classes is not included in the kernel language but may be

considered in a future version.

Axiom:
A ` x :: τ

if τ is a generic instance of A(x)

Application:
A ` e1 :: τ1 → τ2 A ` e2 :: τ1

A ` e1 e2 :: τ2

Abstraction:
A[x/τ ] ` e :: τ ′

A ` \x->e :: τ → τ ′
if τ is a type expression

Existential:
A[x/τ ] ` e :: τ ′

A ` let x free in e :: τ ′
if τ is a type expression

Figure 1: Typing rules for Curry programs

where T is a type constructor of arity n, as defined by a datatype declaration (cf. Section 2.1),8 and
τ1, . . . , τn are type expressions (note that list, tuple and function types have the special syntax [·],
(·,· · ·,·), and ·->· as described above). For instance, [(Int,a)]->a is a type expression containing
the type variable a. A type scheme is a type expression with a universal quantification for some
type variables, i.e., it has the form ∀α1, . . . , αn.τ (n ≥ 0; in case of n = 0, the type scheme is
equivalent to a type expression). A function type declaration f::τ is considered as an assignment
of the type scheme ∀α1, . . . , αn.τ to f , where α1, . . . , αn are all type variables occurring in τ . The
type τ is called a generic instance of the type scheme ∀α1, . . . , αn.τ ′ if there is a substitution
σ = {α1 7→ τ1, . . . , αn 7→ τn} on the types with σ(τ ′) = τ .

The types of all defined functions are collected in a type environment A which is a mapping
from identifiers to type schemes. It contains at least the type schemes of the defined functions and
an assignment of types for some local variables. An expression e is well-typed and has type τ w.r.t.
a type environment A if A ` e :: τ is derivable according to the inference rules shown in Figure 1.

A defining equation f t1 . . . tn = e [where x free] is well-typed w.r.t. a type environment
A if A(f) = ∀α1, . . . , αm.τ [and A(x) is a type expression] and A ` \t1-> · · · \tn->e :: τ is derivable
according to the above inference rules. A conditional equation l | c = r is considered (for the
purpose of typing) as syntactic sugar for l = cond c r where cond is a new function with type
scheme A(cond) = ∀α.Success→ α → α.

A program is well-typed if all its rules are well-typed with a unique assignment of type schemes
to defined functions.9

Note that the following recursive definition is a well-typed Curry program according to the
definition above (and the type definitions given in the prelude, cf. Appendix B):

f :: [a] -> [a]

f x = if length x == 0 then fst (g x x) else x

g :: [a] -> [b] -> ([a],[b])

g x y = (f x, f y)

8We assume here that all type constructors introduced by type synonyms (cf. Section 2.2) are replaced by their

definitions.
9Here we assume that all local declarations are eliminated by the transformations described in Appendix D.7.

21



h :: ([Int],[Bool])

h = g [3,4] [True,False]

However, if the type declaration for g is omitted, the usual type inference algorithms are not able
to infer this type.

5 Expressions

Expressions are a fundamental notion of Curry. As introduced in Section 2, functions are defined
by equations defining expressions that are equivalent to specific function calls. For instance, the
program rule

square x = x*x

defines that the function call (square 3) is equivalent to the expression (3*3).
Expressions occur in conditions and right-hand sides of equations defining functions. A compu-

tation evaluates an expression to a data term (see Section 3). Expressions are built from constants
of a specific data type (e.g., integer constants, character constants, see Section 4), variables, or ap-
plications of constructors or functions to expressions. Furthermore, Curry provides some syntactic
extensions for expressions that are discussed in this section.

5.1 Arithmetic Sequences

Curry supports two syntactic extensions to define list of elements in a compact way. The first one
is a notation for arithmetic sequences. The arithmetic sequence

[ e1 , e2 .. e3 ]

denotes a list of integers starting with the first two elements e1 and e2 and ending with the element
e3 (where e2 and e3 can be omitted). The precise meaning of this notation is defined by the
following translation rules:

Arithmetic sequence notation: Equivalent to:

[e .. ] enumFrom e

[e1,e2 .. ] enumFromThen e1 e2

[e1 .. e2] enumFromTo e1 e2

[e1,e2 .. e3] enumFromThenTo e1 e2 e3

The functions for generating the arithmetic sequences, enumFrom, enumFromThen, enumFromTo, and
enumFromThenTo, are defined in the prelude (see page 62). Thus, the different notations have the
following meaning:

• The sequence [e..] denotes the infinite list [e,e+1,e+2,...].

• The sequence [e1..e2] denotes the finite list [e1,e1+1,e1+2,...,e2]. Note that the list is
empty if e1 > e2.

• The sequence [e1,e2..] denotes the infinite list [e1,e1+i,e1+2*i,...] with i = e2−e1. Note
that i could be positive, negative or zero.

• The sequence [e1,e2..e3] denotes the finite list [e1,e1+i,e1+2*i,...,e3] with i = e2 − e1.
Note that e3 is not contained in this list if there is no integer m with e3 = e1 + m ∗ i.

For instance, [0,2..10] denotes the list [0,2,4,6,8,10].

5.2 List Comprehensions

The second compact notation for lists are list comprehensions. They have the general form

[ e | q1 ,..., qk ]

where k ≥ 1 and each qi is a qualifier that is either

• a generator of the form p <- l, where p is a local pattern (i.e., an expression without defined
function symbols and without multiple occurrences of the same variable, compare Section 2.4)
of type t and l is an expression of type [t], or

• a guard, i.e., an expression of type Bool.

The variables introduced in a local pattern can be used in subsequent qualifiers and the element
description e. Such a list comprehension denotes the list of elements which are the result of
evaluating e in the environment produced by the nested and depth-first evaluation of the generators
satisfying all guards. For instance, the list comprehension

[ x | x <- [1..50], x ‘mod‘ 7 == 0 ]

denotes the list [7,14,21,28,35,42,49], and the list comprehension

[ (x,y) | x<-[1,2,3], y<-[4,5] ]

denotes the list [(1,4),(1,5),(2,4),(2,5),(3,4),(3,5)].
The precise meaning of list comprehensions is defined by the following translation rules (for the

purpose of this definition, we also consider list comprehensions with an empty qualifier list):

List comprehension: Equivalent to:

[e | ] [e]

[e | b, q] if b then [e|q] else [] (b Boolean guard)

[e | let decls, q] let decls in [e|q]

[e | p <- l, q] let ok p = [e|q]
ok p1 = []
...
ok pn = []

in concatMap ok l

where {p1, . . . , pn} = complPat(p)

In the last translation rule, concatMap is defined in the prelude (see page 59) and complPat(p)
denotes the set of patterns that are incompatible with p. This set is defined by

complPat(x) = {}

for all variables x and

complPat(C p1 . . . pn) = {C1 x11 . . . x1n1 , . . . , Ck xk1 . . . xknk
}

∪ {C p1 . . . pi−1 p′ yi+1 . . . yn | 1 ≤ i ≤ n, p′ ∈ complPat(pi)}

where C,C1, . . . , Ck are all the constructors of the result type of C and all xij , yi are fresh variables.
For instance, complPat([True]) = {[], False:zs, True:y:ys} and complPat((x,y)) = {}.

Note that this translation does not work for patterns containing number constants since in this
case the complement set of patterns is infinite. Therefore, we modify this scheme as follows. If the
pattern p in a list comprehension [e | p <- l, q] contains the integer, float or character constants
c1, . . . , ck, we replace them by corresponding fresh variables z1, . . . , zk in the pattern and apply for
this modified pattern p′ the above translation scheme but with the following modified first rule for
ok:

ok p′ = if z1==c1 && ... && zk==ck then [e|q] else []

For instance, the list comprehension

[x | (2,x) <- [(1,3),(2,4),(3,6)]]

(which is evaluated to [4]) is translated into the following expression:

let ok (y,x) = if y==2 then [x] else []

in concatMap ok [(1,3),(2,4),(3,6)]

5.3 Case Expressions

Case expressions are a convenient notation of sequential pattern matching with defaults. The
simplest form of a case expression is as follows (e, e1, . . . , en are expressions and the patterns
p1, . . . , pn are data terms):

case e of

p1 -> e1

. . .

pn -> en

Note that case expressions use the layout rule (see Section C.2). Thus, the patterns p1, . . . , pn

must be vertically aligned. The informal operational meaning of the case expression is as follows.
Evaluate e so that it matches a pattern pi. If this is possible, replace the entire case expression
by the corresponding alternative ei (after replacing the pattern variables occurring in pi by their
actual expressions). If none of the patterns p1, . . . , pn matches, the computation fails. The pattern
matching is tried sequentially, from top to bottom, and rigid, i.e., without binding free variables
occurring in e. In particular, the evaluation of a case expression suspends if the discriminating
expression e evaluates to a free variable.

Case expressions are a convenient notation for functions with default cases. For instance, the
function

swap z = case z of

[x,y] -> [y,x]

_ -> z

returns a list with swapped elements in case of an input list with exactly two elements and is the
identity in all other cases. Moreover, it suspends if the input is a free variable. If we ignore this
latter property, we can define swap by the following rules:

swap [] = []

swap [x] = [x]

swap [x,y] = [y,x]

swap (x1:x2:x3:xs) = x1:x2:x3:xs

The latter definition shows the improved readability obtained by case expressions.
Case expressions might also contain guards and local declarations in the alternatives. For

instance, the following expression is legal:

case y of

Left z

| z >= 0 -> sqr z

| otherwise -> - sqr z

where sqr x = x * x

_ -> 0

Guard expressions must be of type Bool. Case alternatives with guards have a “fall-through”
semantics: if all guards of a case alternative evaluate to False, the match is continued with the
next alternative, i.e., this is handled as if the pattern did not match at all. For instance,

case (1,3) of

(x,y) | x < 0 -> (0,y)

z -> z

returns the pair (1,3).

5.4 Ensuring Instantiation

In some situations one is interested in defining functions that suspend on unknown inputs, e.g., in
concurrent programming (see Appendix A.6). This behavior can be enforced by case expressions
for particular user-defined datatypes. In order to support a similar definition for primitive types
or types with a large number of constructors, there is a predefined primitive function

ensureNotFree :: a -> a

that evaluates the argument head normal form (i.e., to an expression without a defined function at
the top) and suspends if the result is a free variable. The computation proceeds if the free variable
has been instantiated to a non-variable term. In this case, the evaluated argument is returned (i.e.,
semantically it is the identity function).

Based on this function, one can also define more complex operations that evaluate larger parts
of an expression and suspend as long as some parts are instantiated. For instance, the following
function (which is defined in the standard prelude, see Appendix B), evaluates the argument to
spine form, i.e., the skeleton of the list without the elements, and returns it. It suspends if the
result is a non-variable spine.

ensureSpine :: [a] -> [a]

ensureSpine l = ensureList (ensureNotFree l)

where ensureList [] = []

ensureList (x:xs) = x : ensureSpine xs

26



This function is useful to model objects that process list of messages (see Appendix A.6 for an
example).

6 Modules

A module defines a collection of datatypes, constructors and functions which we call entities in the
following. A module exports some of its entities which can be imported and used by other modules.
An entity which is not exported is not accessible from other modules.

A Curry program is a collection of modules. Every module named M (where M does not contain
a dot, see below) must be stored in a file named M.curry. There is one main module which is
loaded into a Curry system. The modules imported by the main module are implicitly loaded but
not visible to the user. After loading the main module, the user can evaluate expressions which
contain entities exported by the main module.

There is one distinguished module, named Prelude, (see Appendix B) which is implicitly im-
ported into each module provided that this module does not contain an explicit import declaration
for the module Prelude. Thus, the entities defined in the standard prelude (basic functions for
arithmetic, list processing etc.) can be always used (if they are not hidden by a hiding declaration,
see below).

A module always starts with the head which contains at least the name of the module followed
by the keyword where, like

module Stack where ...

If a module stored in a directory in file M.curry does not contain a module head, the standard
module head “module M where” is implicitly inserted. Module names can be given a hierarchical
structure by inserting dots which is useful if larger applications should be structured into different
subprojects. For instance,

company.project1.subproject2.mod4

is a valid module name. The dots may reflect the hierarchical file structure where mod-
ules are stored. For instance, the module compiler.symboltable could be stored in the file
symboltable.curry in the directory compiler. To avoid such long module names when refer-
ring to entities of this module by qualification, imported modules can be renamed by the use of
“as” in an import declaration (see below).

Without any further restrictions in the module head, all entities which are defined by top-level
declarations in this module are exported (but not those entities that are imported into this module).
In order to modify this standard set of exported entities of a module, an export list can be added
to the module head. For instance, a module with the head

module Stack(stackType, push, pop, newStack) where ...

exports the entities stackType, push, pop, and newStack. An export list can contain the following
entries:

1. Names of datatypes: This exports only the datatype, whose name must be accessible in this
module, but not the constructors of the datatype. The export of a datatype without its
constructors allows the definition of abstract datatypes.

2. Datatypes with all constructors: If the export list contains the entry t(..), then t must be
a datatype whose name is in the module. In this case, the datatype t and all constructors of

this datatype, which must be also accessible in this module, are exported.

3. Names of functions: This exports the corresponding functions whose names must be accessible
in this module. The types occurring in the argument and result type of this function are
implicitly exported, otherwise the function may not be applicable outside this module.

4. Modules: The set of all accessible entities imported from a module M into the current module
(see below) can be exported by a single entry “module M” in the export list. For instance,
if the head of the module Stack is defined as above, the module head

module Queue(module Stack, enqueue, dequeue) where ...

specifies that the module queue exports the entities stackType, push, pop, newStack,
enqueue, and dequeue.

The unqualified names of the exported entities of a module must be pairwise different to avoid
name clashes in the use of these entities. For instance, the module

module M(f,Ma.g) where

import Ma

f x = Ma.g x

exports the names M.f and M.g, i.e., a qualified entity consists always of the name of the exported
module followed by a dot and the unqualified name of the entity. Spaces are not allowed between
the dot and the module/function name. Thus, an identifier immediately followed by a dot and a
variable or infix operator identifier is always interpreted as a qualified name.

If module Ma also exports the entity f, then the export declaration

module M(f,Ma.f) where

import Ma

f x = Ma.f x

is not allowed since the exported name M.f cannot be uniquely resolved.
All entities defined by top-level declarations in a module are always accessible in this module,

i.e., there is no need to qualify the names of top-level declarations. However, they can be qualified
for the sake of readability, i.e., inside module M, it is always possible to refer to an entity f defined
at the top level of this module with the qualified name M.f. Additionally, the entities exported by
another module can be also made accessible in the module by an import declaration. An import
declaration consists of the name of the imported module and (optionally) a list of entities imported
from that module. If the list of imported entities is omitted, all entities exported by that module
are imported. For instance, the import declaration

import Stack

imports all entities exported by the module Stack, whereas the declaration

import Family(father, grandfather)

imports only the entities father and grandfather from the module Family, provided that they
are exported by Family. Similarly to export declarations, a datatype name t in an import list

imports only the datatype without its constructors whereas the form t(..) imports the datatype
together with all its constructors (provided that they are also exported).

The names of all imported entities are accessible in the current module, i.e., they are equivalent
to top-level declarations, provided that their names are not in conflict with other names. For
instance, if a function f is imported from module M but the current module contains a top-level
declaration for f (which is thus directly accessible in the current module), the imported function is
not accessible (without qualification). Similarly, if two identical names are imported from different
modules and denote different entities, none of these entities is accessible (without qualification). It
is possible to refer to such imported but not directly accessible names by prefixing them with the
module identifier (qualification). For instance, consider the module M1 defined by

module M1 where

f :: Int -> Int

...

and the module m2 defined by

module M2 where

f :: Int -> Int

...

together with the main module

module Main where

import M1

import M2

...

Then the names of the imported functions f are not directly accessible in the main module but one
can refer to the corresponding imported entities by the qualified identifiers M1.f or M2.f.

Another method to resolve name conflicts between imported entities is the qualification of an
imported module. If we change the Main module to

module Main where

import qualified M1

import M2

...

then the name f refers to the entity M2.f since all entities imported from M1 are only accessible by
qualification like M1.f.

A further method to avoid name conflicts is the hiding of imported entities. Consider the
following definition:

module Main where

import M1 hiding (f)

import M2

...

The name f in the Main module refers to the entity M2.f since the name f is not imported from
M1 by the hiding declaration. The hiding clause effects only unqualified names, i.e., the entity

M1.f is still accessible in the body of the Main module. Therefore, a hiding clause has no effect in
combination with a qualified import. Similarly to export declarations, a datatype t in a hiding
clause hides only the datatype (but not its constructors) whereas the form t(..) hides the complete
datatype including its constructors.

The effect of several import declarations is cumulative, i.e., if an entity is hidden in one import
declaration, it can still be imported by another import declaration. For instance, if module Mt

exports a datatype t together with its constructors, then the import declarations

import Mt hiding (t(..))

import Mt (t)

imports all entities exported by Mt but only the name t of the datatype without its constructors,
since the first hiding clause imports everything from Mt except the complete datatype t and the
second import specification additionally imports the name of the datatype t.

Imported modules can also be given a new local name in the import declaration. For instance,
the declaration

import M(f) as foo

enables access to the name f (provided that it is not in conflict with another entity with the same
name) and foo.f but not to M.f. This local renaming enables the abbreviation of long module
names and the substitution of different modules without changing the qualifiers inside a module.

Although each name refers to exactly one entity, it is possible that the same entity is referred
by different names. For instance, consider the modules defined by

module M(f) where

f :: Int -> Int

...

module M1(M.f) where

import M

...

module M2(M.f) where

import M

...

together with the main module

module Main where

import M1

import M2

...

Now the names f, M1.f, and M2.f refer to the identical entity, namely the function f defined
in module M. Note that there is no need to qualify f in the module Main since this name is
unambiguously resolved to the function f defined in module M, although it is imported via two
different paths.

Qualified names are treated syntactically like unqualified names. In particular, a qualified infix
operator like Complex.+ has the same fixity as the definition of + in the module Complex, i.e., the

expression “x Complex.+ y” is syntactically valid. To distinguish between a function composition
like “f . g”, where “.” is an infix operator (see Section B, page 53), and a module qualification,
spaces are not allowed in the module qualification while there must be at least one space before
the “.” if it is used as an infix operator. Thus, “f . g” is interpreted as the composition of f and
g whereas “f.g” is interpreted as the object g imported from module f.

The import dependencies between modules must be non-circular, i.e., it is not allowed that
module m1 imports module m2 and module m2 also imports (directly or indirectly) module m1.

7 Input/Output

Curry provides a declarative model of I/O by considering I/O operations as transformations on
the outside world. In order to avoid dealing with different versions of the outside world, it must
be ensured that at each point of a computation only one version of the world is accessible. This is
guaranteed by using the monadic I/O approach [42] of Haskell and by requiring that I/O operations
are not allowed in program parts where non-deterministic search is possible.

7.1 Monadic I/O

In the monadic I/O approach, the outside “world” is not directly accessible but only through actions
which change the world. Thus, the world is encapsulated in an abstract datatype which provides
actions to change the world. The type of such actions is IO t which is an abbreviation for

World -> (t,World)

where World denotes the type of all states of the outside world. If an action of type IO t is applied
to a particular world, it yields a value of type t and a new (changed) world. For instance, getChar
is an action which reads a character from the standard input when it is executed, i.e., applied to
a world. Therefore, getChar has the type IO Char. The important point is that values of type
World are not accessible for the programmer — she/he can only create and compose actions on
the world. Thus, a program intended to perform I/O operations has a sequence of actions as the
result. These actions are computed and executed when the program is connected to the world by
executing it. For instance,

getChar :: IO Char

getLine :: IO String

are actions which read the next character or the next line from the standard input. The functions

putChar :: Char -> IO ()

putStr :: String -> IO ()

putStrLn :: String -> IO ()

take a character (string) and produces an action which, when applied to a world, puts this character
(string) to the standard output (and a line-feed in case of putStrLn.

Since an interactive program consists of a sequence of I/O operations, where the order in the
sequence is important, there are two operations to compose actions in a particular order. The
function

(>>) :: IO a -> IO b -> IO b

takes two actions as input and yields an action as output. The output action consists of performing
the first action followed by the second action, where the produced value of the first action is ignored.
If the value of the first action should be taken into account before performing the second action,
the actions can be composed by the function

(>>=) :: IO a -> (a -> IO b) -> IO b

where the second argument is a function taking the value produced by the first action as input and

performs another action. For instance, the action

getLine >>= putStrLn

is of type IO () and copies, when executed, a line from standard input to standard output.
The return function

return :: a -> IO a

is sometimes useful to terminate a sequence of actions and return a result of an I/O operation.
Thus, return v is an action which does not change the world and returns the value v.

To execute an action, it must be the main expression in a program, i.e., interactive programs
have type IO (). Since the world cannot be copied (note that the world contains at least the
complete file system), non-determinism in relation with I/O operations must be avoided. Thus, the
applied action must always be known, i.e., >> and >>= suspend if the arguments are free variables.
Moreover, it is a runtime error if a disjunctive expression (cf. Section 3) σ1 e1 |· · ·| σn en, where
the ei’s are of type IO () and n > 1, occurs as the top-level expression of a program, since it is
unclear in this case which of the disjunctive actions should be applied to the current world. Thus,
all possible search must be encapsulated between I/O operations (see Section 8). The compiler
may warn the user about non-deterministic computations which may occur in I/O actions so that
the programmer can encapsulate them.

7.2 Do Notation

To provide a more conventional notation for programming sequences of I/O operations, Curry has
a special piece of syntax for writing such sequences. For instance, consider the following expression
to read a line form the standard input and to print them together with the string "Your input: "

on the standard output:

getLine >>= \line -> putStr "Your input: " >> putStrLn line

Using the do notation, this expression can be written in a more traditional style:

do line <- getLine

putStr "Your input: "

putStrLn line

Note that the do notation is just another syntax for sequences of I/O actions. Thus, do expressions
can be used wherever an expression of type I/O is required (but note that do expressions use the
layout rule (see Section C.2) and, therefore, they are usually written in separate program lines).
The following table specifies the translation of do expressions into the kernel language:

Do notation: Equivalent to:

do expr expr

do expr
stmts

expr >> do stmts

do p <- expr
stmts

expr >>= \p -> do stmts

do let decls
stmts

let decls
in do stmts

8 Encapsulated Search

Global search, possibly implemented by backtracking, must be avoided in some situations (user-
control of efficiency, concurrent computations, non-backtrackable I/O). Hence it is sometimes nec-
essary to encapsulate search, i.e., non-deterministic computations in parts of larger programs.
Non-deterministic computations might occur in Curry whenever a function must be evaluated with
a free variable at a flexible argument position. In this case, the computation must follow different
branches with different bindings applied to the current expression which has the effect that the
entire expression is split into (at least) two independent disjunctions. To give the programmer con-
trol on the actions taken in this situation, Curry provides a primitive search operator which is the
basis to implement sophisticated search strategies. This section sketches the idea of encapsulating
search in Curry and describes some predefined search strategies.

8.1 Search Goals and Search Operators

Since search is used to find solutions to some constraint, search is always initiated by a constraint
containing a search variable for which a solution should be computed.10 Since the search vari-
able may be bound to different solutions in different disjunctive branches, it must be abstracted.
Therefore, a search goal has the type a->Success where a is the type of the values which we are
searching for. In particular, if c is a constraint containing a variable x and we are interested in
solutions for x, i.e., values for x such that c is satisfied, then the corresponding search goal has the
form (\x->c). However, any other expression of the same type can also be used as a search goal.

To control the search performed to find solutions to search goals, Curry has a predefined operator

try :: (a->Success) -> [a->Success]

which takes a search goal and produces a list of search goals. The search operator try attempts to
evaluate the search goal until the computation finishes or performs a non-deterministic splitting.
In the latter case, the computation is stopped and the different search goals caused by this splitting
are returned as the result list. Thus, an expression of the form try (\x->c) can have three possible
results:

1. An empty list. This indicates that the search goal (\x->c) has no solution. For instance, the
expression

try (\x -> 1=:=2)

reduces to [].

2. A list containing a single element. In this case, the search goal (\x->c) has a single solution
represented by the element of this list. For instance, the expression

try (\x->[x]=:=[0])

reduces to [\x->x=:=0]. Note that a solution, i.e., a binding for the search variable like the
substitution {x 7→ 0}, can always be presented by an equational constraint like x=:=0.

10The generalization to more than one search variable is straightforward by the use of tuples.

Generally, a one-element list as a result of try has always the form [\x->x=:=e] (plus some
local variables, see next subsection) where e is fully evaluated, i.e., e does not contain defined
functions. Otherwise, this goal might not be solvable due to the definition of equational
constraints.

3. A list containing more than one element. In this case, the evaluation of the search goal
(\x->c) requires a non-deterministic computation step. The different alternatives immedi-
ately after this non-deterministic step are represented as elements of this list. For instance,
if the function f is defined by

f a = c

f b = d

then the expression

try (\x -> f x =:= d)

reduces to the list [\x->x=:=a & f a =:= d, \x->x=:=b & f b =:= d]. This example also
shows why the search variable must be abstracted: the alternative bindings cannot be actually
performed (since a free variable is only bound to at most one value in each computation
thread) but are represented as equational constraints in the search goal.

Note that the search goals of the list in the last example are not further evaluated. This provides
the possibility to determine the behavior for non-deterministic computations. For instance, the
following function defines a depth-first search operator which collects all solutions of a search goal
in a list:

solveAll :: (a->Success) -> [a->Success]

solveAll g = collect (try g)

where collect [] = []

collect [g] = [g]

collect (g1:g2:gs) = concat (map solveAll (g1:g2:gs))

(concat concatenates a list of lists to a single list). For instance, if append is the list concatenation,
then the expression

solveAll (\l -> append l [1] =:= [0,1])

reduces to [\l->l=:=[0]].
The value computed for the search variable in a search goal can be easily accessed by applying

it to a free variable. For instance, the evaluation of the applicative expression

solveAll (\l->append l [1] =:= [0,1]) =:= [g] & g x

binds the variable g to the search goal [\l->l=:=[0]] and the variable x to the value [0] (due to
solving the constraint x=:=[0]). Based on this idea, there is a predefined function

findall :: (a->Success) -> [a]

which takes a search goal and collects all solutions (computed by a depth-first search like solveAll)
for the search variable into a list.

Due to the laziness of the language, search goals with infinitely many solutions cause no problems
if one is interested only in finitely many solutions. For instance, a function which computes only
the first solution w.r.t. a depth-first search strategy can be defined by

first g = head (findall g)

Note that first is a partial function, i.e., it is undefined if g has no solution.

8.2 Local Variables

Some care is necessary if free variables occur in the search goal, like in the goal

\l2 -> append l1 l2 =:= [0,1]

Here, only the variable l2 is abstracted in the search goal but l1 is free. Since non-deterministic
bindings cannot be performed during encapsulated search, free variables are never bound inside en-
capsulated search. Thus, if it is necessary to bind a free variable in order to proceed an encapsulated
search operation, the computation suspends. For instance, the expression

first (\l2 -> append l1 l2 =:= [0,1])

cannot be evaluated and will be suspended until the variable l1 is bound by another part of the
computation. Thus, the constraint

s =:= first (\l2->append l1 l2 =:= [0,1]) & l1 =:= [0]

can be evaluated since the free variable l1 in the search goal is bound to [0], i.e., this constraint
reduces to the answer

{l1=[0], s=[1]}

In some cases it is reasonable to have unbound variables in search goals, but these variables should
be treated as local, i.e., they might have different bindings in different branches of the search. For
instance, if we want to compute the last element of the list [3,4,5] based on append, we could try
to solve the search goal

\e -> append l [e] =:= [3,4,5]

However, the evaluation of this goal suspends due to the necessary binding of the free variable l.
This can be avoided by declaring the variable l as local to the constraint by the use of let (see
Section 2.5), like in the following expression:

first (\e -> let l free in append l [e] =:= [3,4,5])

Due to the local declaration of the variable l (which corresponds logically to an existential quan-
tification), the variable l is only visible inside the constraint and, therefore, it can be bound to
different values in different branches. Hence this expression evaluates to the result 5.

In order to ensure that an encapsulated search will not be suspended due to necessary bindings
of free variables, the search goal should be a closed expression when a search operator is applied
to it, i.e., the search variable is bound by the lambda abstraction and all other free variables are
existentially quantified by local declarations.

8.3 Predefined Search Operators

There are a number of search operators which are predefined in the prelude. All these operators are
based on the primitive try as described above. It is also possible to define other search strategies
in a similar way. Thus, the try operator is a a powerful primitive to define appropriate search
strategies. In the following, we list the predefined search operators.

solveAll :: (a->Success) -> [a->Success]

Compute all solutions for a search goal via a depth-first search strategy. If there is no solution
and the search space is finite, the result is the empty list, otherwise the list contains solved
search goals (i.e., without defined operations).

once :: (a->Success) -> (a->Success)

Compute the first solution for a search goal via a depth-first search strategy. Note that once
is partially defined, i.e., if there is no solution and the search space is finite, the result is
undefined.

findall :: (a->Success) -> [a]

Compute all solutions for a search goal via a depth-first search strategy and unpack the
solution’s values for the search variable into a list.

best :: (a->Success) -> (a->a->Bool) -> [a->Success]

Compute the best solution via a depth-first search strategy, according to a specified rela-
tion (the second argument) that can always take a decision which of two solutions is better
(the relation should deliver True if the first argument is a better solution than the second
argument).

As a trivial example, consider the relation shorter defined by

shorter l1 l2 = length l1 <= length l2

Then the expression

best (\x -> let y free in append x y =:= [1,2,3]) shorter

computes the shortest list which can be obtained by splitting the list [1,2,3] into this list
and some other list, i.e., it reduces to [\x->x=:=[]]. Similarly, the expression

best (\x -> let y free in append x y =:= [1,2,3])

(\l1 l2 -> length l1 > length l2)

reduces to [\x->x=:=[1,2,3]].

browse :: (a->Success) -> IO ()

Show the solution of a solved constraint on the standard output, i.e., a call browse g, where
g is a solved search goal, is evaluated to an I/O action which prints the solution. If g is not
an abstraction of a solved constraint, a call browse g produces a runtime error.

browseList :: [a->Success] -> IO ()

Similar to browse but shows a list of solutions on the standard output. The browse operators
are useful for testing search strategies. For instance, the evaluation of the expression

browseList (solveAll (\x -> let y free in append x y =:= [0,1,2]))

produces the following result on the standard output:

[]

[0]

[0,1]

[0,1,2]

Due to the laziness of the evaluation strategy, one can also browse the solutions of a goal
with infinitely many solutions which are printed on the standard output until the process is
stopped.

9 Interface to External Functions and Constraint Solvers

Since Curry has only a small number of builtins, it provides a simple interface to connect external
functions (functions programmed in another language) and external constraint solvers. External
functions must be free of side-effects in order to be compatible with Curry’s computation model.
An external constraint solver consists of a constraint store which can be accessed and manipulated
via a few primitive operations.

9.1 External Functions

External functions are considered as an implementation of a potentially infinite set of equations
(describing the graph of the functions). In particular, they have no side effects, i.e., identical
function calls at different points of time yield identical results. Since the implementation of an
external function is unknown, a call to an external function is suspended until all arguments are
fully known, i.e., until they are evaluated to ground data terms. This view is compatible with the
residuation principle covered by Curry’s computation model and very similar to the connection of
external functions to logic programs [10, 11].

An external function is declared by a type declaration followed by external(FName,OF,Lang)

where FName is the name of the external function and its code is contained in the object file OF.
Lang is the implementation language of the external function. For instance, if the addition and
multiplication on integers are defined as C functions named add and mult where the compiled code
is contained in the file “arith.o”, we can connect these functions to a Curry program by providing
the declarations

add :: Int -> Int -> Int external("add","arith.o",C)

mult :: Int -> Int -> Int external("mult","arith.o",C)

Implementations of Curry might have restrictions on the interface. A reasonable requirement is
that the implementation language is constrained to C and the argument and result types are only
simple types like Bool, Int, Float, Char, or String.

9.2 External Constraint Solvers

A constraint solver can be viewed as an abstract datatype consisting of a constraint store together
with a few operations to check the entailment of a constraint or to add a new constraint. In order
to connect a constraint solver to Curry, like a solver for arithmetic or Boolean constraints, the
external solver must provide the following operations (cs denotes the type of the constraint store
and c the type of constraints):

new : → cs (create and initialize a new constraint store)
ask : cs× c → {True, False, Unknown} (check entailment of a constraint)
tell : cs× c → cs (add a new constraint)
clone : cs → cs× cs (copy the constraint store)

Using these operations, it is relatively easy to extend Curry’s computation model to include con-
straints by adding the constraint store as a new component to the substitution part in answer

expressions (cf. Section 3).11

11This section will be modified or extended in a later version.

10 Literate Programming

To encourage the use of comments for the documentation of programs, Curry supports a literate
programming style (inspired by Donald Knuth’s “literate programming”) where comments are the
default case. Non-comment lines containing program code must start with “>” followed by a blank.
Using this style, we can define the concatenation and reverse function on lists by the following
literate program:

The following function defines the concatenation of two lists.

Note that this function is predefined as ‘++’ in the standard prelude.

> append :: [t] -> [t] -> [t]

> append [] x = x

> append (x:xs) ys = x : append xs ys

As a second example for list operations, we define a

function to reverse the order of elements in a list:

> rev :: [t] -> [t]

> rev [] = []

> rev (x:xs) = append (rev xs) [x]

To distinguish literature from non-literate programs, literate programs must be stored in files
with the extension “.lcurry” whereas non-literate programs are stored in files with the extension
“.curry”.

11 Interactive Programming Environment

In order to support different implementations with a comparable user interface, the following com-
mands should be supported by each interactive programming environment for Curry (these com-
mands can also be abbreviated to :c where c is an unambiguous prefix string of the command):

:load file Load the program stored in file.curry.

:reload Repeat the last load command.

expr Evaluate the expression expr and show all computed results. Since an expression could
be evaluated to a disjunctive expression (cf. Section 3), the expression could be automati-
cally wrapped in some search operator like depth-first search or a fair breadth-first search,
depending on the implementation.

:debug expr Debug the evaluation of the expression expr, i.e., show the single computation steps
and ask the user what to do after each single step (like proceed, abort, etc.).

:type expr Show the type of the expression expr.

:eval f Show the definitional tree (see Section D.1) of function f.

:quit Exit the system.

A Example Programs

This section contains a few example programs together with some initial expressions and their
computed results.

A.1 Operations on Lists

Here are simple operations on lists. Note that, due to the logic programming features of Curry,
append can be used to split lists. We exploit this property to define the last element of a list in a
very simple way.

-- Concatenation of two lists:

append :: [t] -> [t] -> [t]

append [] ys = ys

append (x:xs) ys = x:append xs ys

-- Naive reverse of all list elements:

rev :: [t] -> [t]

rev [] = []

rev (x:xs) = append (rev xs) [x]

-- Last element of a list:

last :: [t] -> t

last xs | append ys [x] =:= xs = x where x,ys free

Expressions and their evaluated results:

append [0,1] [2,3] ; [0,1,2,3]

append L M =:= [0,1]

; {L=[],M=[0,1]} | {L=[0],M=[1]} | {L=[0,1],M=[]}

rev [0,1,2,3] ; [3,2,1,0]

last (append [1,2] [3,4]) ; 4

A.2 Higher-Order Functions

Here are some “traditional” higher-order functions to show that the familiar functional program-
ming techniques can be applied in Curry. Note that the functions map, foldr, and filter are
predefined in Curry (see Appendix B).

-- Map a function on a list (predefined):

map :: (t1->t2) -> [t1] -> [t2]

map f [] = []

map f (x:xs) = f x:map f xs

-- Fold a list (predefined):

foldr :: (t1->t2->t2) -> t2 -> [t1] -> t2

foldr f a [] = a

foldr f a (x:xs) = f x (foldr f a xs)

-- Filter elements in a list (predefined):

filter :: (t -> Bool) -> [t] -> [t]

filter p [] = []

filter p (x:xs) = if p x then x:filter p xs

else filter p xs

-- Quicksort function:

quicksort :: [Int] -> [Int]

quicksort [] = []

quicksort (x:xs) = quicksort (filter (<= x) xs)

++ [x]

++ quicksort (filter (> x) xs)

A.3 Relational Programming

Here is a traditional example from logic programming: a simple deductive database with family
relationships. We use a relational programming style, i.e., all relationships are represented as
constraints (i.e., functions with result type Success).

-- Declaration of an enumeration type for persons:

-- (as an alternative, one could consider persons as strings)

data Person = Christine | Maria | Monica | Alice | Susan |

Antony | Bill | John | Frank | Peter | Andrew

-- Two basic relationships:

married :: Person -> Person -> Success

married Christine Antony = success

married Maria Bill = success

married Monica John = success

married Alice Frank = success

mother :: Person -> Person -> Success

mother Christine John = success

mother Christine Alice = success

mother Maria Frank = success

mother Monica Susan = success

mother Monica Peter = success

mother Alice Andrew = success

-- and here are the deduced relationships:

father :: Person -> Person -> Success

father f c = let m free in married m f & mother m c

grandfather :: Person -> Person -> Success

grandfather g c = let f free in father g f & father f c

grandfather g c = let m free in father g m & mother m c

Expressions and their evaluated results:

father John child ; {child=Susan} | {child=Peter}

grandfather g c ;

{g=Antony,c=Susan} | {g=Antony,c=Peter} |

{g=Bill,c=Andrew} | {g=Antony,c=Andrew}

A.4 Functional Logic Programming

This is the same example as in the previous section. However, we use here a functional logic
programming style which is more readable but provides the same goal solving capabilities. The
basic functions are husband and mother which express the functional dependencies between the
different persons. Note that the derived function grandfather is a non-deterministic function
which yields all grandfathers for a given person.

data Person = Christine | Maria | Monica | Alice | Susan |

Antony | Bill | John | Frank | Peter | Andrew

-- Two basic functional dependencies:

husband :: Person -> Person

husband Christine = Antony

husband Maria = Bill

husband Monica = John

husband Alice = Frank

mother :: Person -> Person

mother John = Christine

mother Alice = Christine

mother Frank = Maria

mother Susan = Monica

mother Peter = Monica

mother Andrew = Alice

-- and here are the deduced functions and relationships:

father :: Person -> Person

father c = husband (mother c)

grandfather :: Person -> Person

grandfather g = father (father g)

grandfather g = father (mother g)

Expressions and their evaluated results:

father Child =:= john ; {Child=susan} | {Child=peter}

grandfather c ;

{c=Susan} Antony | {c=Peter} Antony | {c=Andrew} Bill | {c=Andrew} Antony

A.5 Constraint Solving and Concurrent Programming

In this example we demonstrate how Curry’s concurrent features can be used to solve constraint
problems with finite domains more efficiently than a simple generate-and-test solver. We want to
solve the classical map coloring problem. Consider the following simple map:

Country 1

Country 2

Country 3

Country 4

The problem is to assign to each of the four countries a color (red, green, yellow, or blue) such that
countries with a common border have different colors.

To solve this problem, we have to specify that some countries have different colors. For this
purpose, we define the following constraint which checks to elements for incompatibility:

diff :: a -> a -> Success

diff x y = (x==y)=:=False

For instance, diff t1 t2 is satisfied if t1 and t2 are different constants. If one of the arguments is a
variable, the evaluation of the function is delayed by definition of ==.

Now there is a straightforward solution of the map coloring problem. We define a constraint
coloring specifying the valid colors for each country and a constraint correct specifying which
countries must have different colors:

data Color = Red | Green | Yellow | Blue

isColor :: Color -> Success

isColor Red = success

isColor Yellow = success

isColor Green = success

isColor Blue = success

coloring :: Color -> Color -> Color -> Color -> Success

coloring l1 l2 l3 l4 = isColor l1 & isColor l2 & isColor l3 & isColor l4

correct :: Color -> Color -> Color -> Color -> Success

correct l1 l2 l3 l4 =

diff l1 l2 & diff l1 l3 & diff l2 l3 & diff l2 l4 & diff l3 l4

In classical logic programming, we can compute the solutions to the map coloring problem by
enumerating all potential solutions followed by a check whether a potential solution is a correct

one (“generate and test”). This can be expressed by solving the following goal:12

coloring l1 l2 l3 l4 & correct l1 l2 l3 l4

However, a much faster solution can be obtained by reversing the order of the tester and the
generator:

correct l1 l2 l3 l4 & coloring l1 l2 l3 l4

The latter constraint is evaluated in a concurrent way. In the first steps, the subexpression
correct l1 l2 l3 l4 is reduced to the constraint

diff l1 l2 & diff l1 l3 & diff l2 l3 & diff l2 l4 & diff l3 l4

which is then reduced to

(l1==l2)=:=False & (l1==l3)=:=False & (l2==l3)=:=False & (l2==l4)=:=False

& (l3==l4)=:=False

This constraint cannot be further evaluated since the arguments to == are free variables. Therefore,
it is suspended and the final equational constraint coloring l1 l2 l3 l4 is evaluated which binds
the variables to the potential colors. Since the variable binding is performed by consecutive compu-
tation steps, the equalities are evaluated as soon as their arguments are bound. For instance, if the
variables l1 and l2 are bound to the color Red, the first constraint (l1==l2)=:=False cannot be
solved (due to the unsolvability of the equational constraint True=:=False) which causes the failure
of the entire goal. As a consequence, not all potential colorings are computed (as in the generate-
and-test approach) but only those colorings for which the constraints correct l1 l2 l3 l4 is
satisfiable. Therefore, the (dis)equality goals act as “passive constraints” aiming to reduce the
search space.

12We assume an implementation that processes the concurrent conjunction c1 & c2 in a “sequential-first” manner:

first c1 is solved and, if this is not possible due to a suspended evaluation, c2 is solved.

A.6 Concurrent Object-Oriented Programming

The following example shows a simple method to program in a concurrent object-oriented style in
Curry. For this purpose, an object is a process waiting for incoming messages. The local state is a
parameter of this process. Thus, a process is a function of type

State -> [Message] -> Success

In the subsequent example, we implement a bank account as an object waiting for messages of the
form Deposit i, Withdraw i, or Balance i. Thus, the bank account can be defined as follows:

data Message = Deposit Int | Withdraw Int | Balance Int

account :: Int -> [Message] -> Success

account _ [] = success

account n (Deposit a : ms) = account (n+a) ms

account n (Withdraw a : ms) = account (n-a) ms

account n (Balance b : ms) = b=:=n & account n ms

-- Install an initial account with message stream s:

make_account s = account 0 (ensureSpine s)

A new account object is created by the constraint “make_account s” where s is a free variable.
The function ensureSpine (see Section 5.4) used in the definition of make_account ensures that
the evaluation of the function call to account suspends on uninstantiated parts of the message list

When s is instantiated with messages, the account object starts processing these messages. The
following concurrent conjunction of constraints creates an account and sends messages:

make_account s & s=:=[Deposit 200, Deposit 50, Balance b]

After this goal is solved, the free variable b has been bound to 250 representing the balance after
the two deposits.

To show a client-server interaction, we define a client of a bank account who is finished if his
account is 50, buys things for 30 if his account is greater than 50, and works for an income of
70 if his account is less than 50. In order to get the client program independent of the account
processing, the client sends messages to his account. Therefore, the client is implemented as follows:

-- Send a message to an object identified by its message stream obj:

sendMsg :: [msg] -> msg -> [msg]

sendMsg msg obj | obj =:= msg:obj1 = obj1 where obj1 free

-- Client process of an bank account

client s | s1 =:= sendMsg (Balance b) s =

if b==50 then s1=:=[] -- stop

else if b>50 then client (sendMsg (Withdraw 30) s1) -- buy

else client (sendMsg (Deposit 70) s1) -- work

where s1,b free

We start the account and client process with an initial amount of 100:

make_account s & client (sendMsg (Deposit 100) s)

A Curry system evaluates this goal by a concurrent evaluation of both processes and computes the
final answer

{s=[Deposit 100, Balance 100, Withdraw 30, Balance 70, Withdraw 30,

Balance 40, Deposit 70, Balance 110, Withdraw 30, Balance 80,

Withdraw 30, Balance 50]}

which shows the different messages sent to the account process.

B Standard Prelude

This section contains the standard prelude for Curry programs. The module Prelude will be
imported into every Curry module unless the module has an explicit import declaration for the
module Prelude (for instance, in order to hide some names from the standard prelude). Note that
some of the definitions shown below are syntactically not allowed (e.g., datatypes for tuples). They
are only provided for documentation purposes.

module Prelude where

-- Infix operator declarations

infixl 9 !!

infixr 9 .

infixl 7 *, /, ‘div‘, ‘mod‘

infixl 6 +, -

infixr 5 ++, :

infix 4 =:=, ==, /=, <, >, <=, >=

infix 4 ‘elem‘, ‘notElem‘

infixr 3 &&

infixr 2 ||

infixl 1 >>, >>=

infixr 0 $, $!, $!!, $#, $##, ‘seq‘, &, &>, ?

-- Some standard combinators:

-- Function composition

(.) :: (b -> c) -> (a -> b) -> (a -> c)

f . g = \x -> f (g x)

-- Identity

id :: a -> a

id x = x

-- Constant function

const :: a -> b -> a

const x _ = x

-- Convert an uncurried function to a curried function

curry :: ((a,b) -> c) -> a -> b -> c

curry f a b = f (a,b)

-- Convert an curried function to a function on pairs

uncurry :: (a -> b -> c) -> (a,b) -> c

uncurry f (a,b) = f a b

-- (flip f) is identical to f but with the order of arguments reversed

flip :: (a -> b -> c) -> b -> a -> c

flip f x y = f y x

-- Repeat application of a function until a predicate holds

until :: (a -> Bool) -> (a -> a) -> a -> a

until p f x = if p x then x else until p f (f x)

-- Evaluate the first argument to head normal form and return the

-- second argument.

seq :: a -> b -> b

-- Evaluate the argument to head normal form and returns it.

-- Suspend until the result is bound to a non-variable term.

ensureNotFree :: a -> a

--- Evaluate the argument to spine form and returns it.

--- Suspend until the result is bound to a non-variable spine.

ensureSpine :: [a] -> [a]

ensureSpine l = ensureList (ensureNotFree l)

where ensureList [] = []

ensureList (x:xs) = x : ensureSpine xs

-- Right-associative application

($) :: (a -> b) -> a -> b

f $ x = f x

-- Right-associative application with strict evaluation of its argument.

($!) :: (a -> b) -> a -> b

f $! x = x ‘seq‘ f x

-- Right-associative application with strict evaluation of its argument

-- to normal form.

($!!) :: (a -> b) -> a -> b

f $!! x | x=:=y = f y where y free

-- Right-associative application with strict evaluation of its argument

-- to a non-variable term.

($#) :: (a -> b) -> a -> b

f $# x = f $! (ensureNotFree x)

-- Right-associative application with strict evaluation of its argument

-- to ground normal form.

($##) :: (a -> b) -> a -> b

f $## x | x=:=y = y==y ‘seq‘ f y where y free

-- Abort the execution with an error message

error :: String -> _

-- failed is a non-reducible polymorphic function.

-- It is useful to express a failure in a search branch of the execution.

failed :: _

failed | 1=:=2 = x where x free

-- Boolean values

data Bool = False | True

-- Sequential conjunction

(&&) :: Bool -> Bool -> Bool

True && x = x

False && _ = False

-- Sequential disjunction

(||) :: Bool -> Bool -> Bool

True || _ = True

False || x = x

-- Negation

not :: Bool -> Bool

not True = False

not False = True

-- Conditional

if_then_else :: Bool -> a -> a -> a

if_then_else b t f = case b of True -> t

False -> f

otherwise :: Bool

otherwise = True

-- Equality on finite ground data terms

(==) :: a -> a -> Bool

-- Disequality

(/=) :: a -> a -> Bool

x /= y = not (x==y)

-- Ordering

data Ordering = LT | EQ | GT

-- Comparison of arbitrary ground data terms.

-- Data constructors are compared in the order of their definition

-- in the datatype declarations and recursively in the arguments.

compare :: a -> a -> Ordering

-- Less-than on ground data terms

(<) :: a -> a -> Bool

x < y = case compare x y of LT -> True

_ -> False

-- Greater-than on ground data terms

(>) :: a -> a -> Bool

x > y = case compare x y of GT -> True

_ -> False

-- Less-or-equal on ground data terms

(<=) :: a -> a -> Bool

x <= y = not (x > y)

-- Greater-or-equal on ground data terms

(>=) :: a -> a -> Bool

x >= y = not (x < y)

-- Maximum of ground data terms

max :: a -> a -> a

max x y = if x >= y then x else y

-- Minimum of ground data terms

min :: a -> a -> a

min x y = if x <= y then x else y

-- Pairs

data (a,b) = (a,b)

fst :: (a,b) -> a

fst (x,_) = x

snd :: (a,b) -> b

snd (_,y) = y

-- Unit type

data () = ()

-- Lists

data [a] = [] | a : [a]

type String = [Char]

-- First element of a list

head :: [a] -> a

head (x:_) = x

-- Remaining elements of a list

tail :: [a] -> [a]

tail (_:xs) = xs

-- Is a list empty?

null :: [_] -> Bool

null [] = True

null (_:_) = False

-- Concatenation

(++) :: [a] -> [a] -> [a]

[] ++ ys = ys

(x:xs) ++ ys = x : xs++ys

-- List length

length :: [a] -> Int

length [] = 0

length (_:xs) = 1 + length xs

-- List index (subscript) operator, head has index 0

(!!) :: [a] -> Int -> a

(x:xs) !! n | n==0 = x

| n>0 = xs !! (n-1)

-- Map a function on a list

map :: (a->b) -> [a] -> [b]

map _ [] = []

map f (x:xs) = f x : map f xs

-- Accumulate all list elements by applying a binary operator from

-- left to right, i.e.,

-- foldl f z [x1,x2,...,xn] = (...((z ‘f‘ x1) ‘f‘ x2) ...) ‘f‘ xn :

foldl :: (a -> b -> a) -> a -> [b] -> a

foldl _ z [] = z

foldl f z (x:xs) = foldl f (f z x) xs

-- Accumulate a non-empty list from left to right:

foldl1 :: (a -> a -> a) -> [a] -> a

foldl1 f (x:xs) = foldl f x xs

-- Accumulate all list elements by applying a binary operator from

-- right to left, i.e.,

-- foldr f z [x1,x2,...,xn] = (x1 ‘f‘ (x2 ‘f‘ ... (xn ‘f‘ z)...)) :

foldr :: (a->b->b) -> b -> [a] -> b

foldr _ z [] = z

foldr f z (x:xs) = f x (foldr f z xs)

-- Accumulate a non-empty list from right to left:

foldr1 :: (a -> a -> a) -> [a] -> a

foldr1 _ [x] = x

foldr1 f (x1:x2:xs) = f x1 (foldr1 f (x2:xs))

-- Filter elements in a list

filter :: (a -> Bool) -> [a] -> [a]

filter _ [] = []

filter p (x:xs) = if p x then x : filter p xs

else filter p xs

-- Join two lists into one list of pairs. If one input list is shorter than

-- the other, the additional elements of the longer list are discarded.

zip :: [a] -> [b] -> [(a,b)]

zip [] _ = []

zip (_:_) [] = []

zip (x:xs) (y:ys) = (x,y) : zip xs ys

-- Join three lists into one list of triples. If one input list is shorter than

-- the other, the additional elements of the longer lists are discarded.

zip3 :: [a] -> [b] -> [c] -> [(a,b,c)]

zip3 [] _ _ = []

zip3 (_:_) [] _ = []

zip3 (_:_) (_:_) [] = []

zip3 (x:xs) (y:ys) (z:zs) = (x,y,z) : zip3 xs ys zs

-- Join two lists into one list by applying a combination function to

-- corresponding pairs of elements (i.e., zip = zipWith (,)):

zipWith :: (a->b->c) -> [a] -> [b] -> [c]

zipWith _ [] _ = []

zipWith _ (_:_) [] = []

zipWith f (x:xs) (y:ys) = f x y : zipWith f xs ys

-- Join three lists into one list by applying a combination function to

-- corresponding triples of elements (i.e., zip3 = zipWith3 (,,)):

zipWith3 :: (a->b->c->d) -> [a] -> [b] -> [c] -> [d]

zipWith3 _ [] _ _ = []

zipWith3 _ (_:_) [] _ = []

zipWith3 _ (_:_) (_:_) [] = []

zipWith3 f (x:xs) (y:ys) (z:zs) = f x y z : zipWith3 f xs ys zs

-- Transform a list of pairs into a pair of lists

unzip :: [(a,b)] -> ([a],[b])

unzip [] = ([],[])

unzip ((x,y):ps) = (x:xs,y:ys) where (xs,ys) = unzip ps

-- Transform a list of triples into a triple of lists

unzip3 :: [(a,b,c)] -> ([a],[b],[c])

unzip3 [] = ([],[],[])

unzip3 ((x,y,z):ts) = (x:xs,y:ys,z:zs) where (xs,ys,zs) = unzip3 ts

-- Concatenate a list of lists into one list

concat :: [[a]] -> [a]

concat l = foldr (++) [] l

-- Map a function from elements to lists and merge the result into one list

concatMap :: (a -> [b]) -> [a] -> [b]

concatMap f = concat . map f

-- Infinite list of repeated applications of a function f to an element x:

-- iterate f x = [x, f x, f (f x),...]

iterate :: (a -> a) -> a -> [a]

iterate f x = x : iterate f (f x)

-- Infinite list where all elements have the same value x:

repeat :: a -> [a]

repeat x = x : repeat x

-- List of length n where all elements have the same value x:

replicate :: Int -> a -> [a]

replicate n x = take n (repeat x)

-- Return prefix of length n

take :: Int -> [a] -> [a]

take n l = if n<=0 then [] else takep n l

where takep _ [] = []

takep n (x:xs) = x : take (n-1) xs

-- Return suffix without first n elements

drop :: Int -> [a] -> [a]

drop n l = if n<=0 then l else dropp n l

where dropp _ [] = []

dropp n (_:xs) = drop (n-1) xs

-- (splitAt n xs) is equivalent to (take n xs, drop n xs)

splitAt :: Int -> [a] -> ([a],[a])

splitAt n l = if n<=0 then ([],l) else splitAtp n l

where splitAtp _ [] = ([],[])

splitAtp n (x:xs) = let (ys,zs) = splitAt (n-1) xs in (x:ys,zs)

-- Return longest prefix with elements satisfying a predicate

takeWhile :: (a -> Bool) -> [a] -> [a]

takeWhile _ [] = []

takeWhile p (x:xs) = if p x then x : takeWhile p xs else []

-- Return suffix without takeWhile prefix

dropWhile :: (a -> Bool) -> [a] -> [a]

dropWhile _ [] = []

dropWhile p (x:xs) = if p x then dropWhile p xs else x:xs

-- (span p xs) is equivalent to (takeWhile p xs, dropWhile p xs)

span :: (a -> Bool) -> [a] -> ([a],[a])

span _ [] = ([],[])

span p (x:xs)

| p x = let (ys,zs) = span p xs in (x:ys, zs)

| otherwise = ([],x:xs)

-- (break p xs) is equivalent to (takeWhile (not.p) xs, dropWhile (not.p) xs)

-- i.e., it breaks a list at the first occurrence of an element satisfying p

break :: (a -> Bool) -> [a] -> ([a],[a])

break p = span (not . p)

-- Break a string into list of lines where a line is terminated at a

-- newline character. The resulting lines do not contain newline characters.

lines :: String -> [String]

lines [] = []

lines (c:cs) = let (l,restcs) = splitline (c:cs) in l : lines restcs

where splitline [] = ([],[])

splitline (c:cs) = if c==’\n’

then ([],cs)

else let (ds,es) = splitline cs in (c:ds,es)

-- Concatenate a list of strings with terminating newlines

unlines :: [String] -> String

unlines ls = concatMap (++"\n") ls

-- Break a string into a list of words where the words are delimited by

-- white spaces.

words :: String -> [String]

words s = let s1 = dropWhile isSpace s

in if s1=="" then []

else let (w,s2) = break isSpace s1

in w : words s2

where

isSpace c = c == ’ ’ || c == ’\t’ || c == ’\n’ || c == ’\r’

-- Concatenate a list of strings with a blank between two strings.

unwords :: [String] -> String

unwords ws = if ws==[] then []

else foldr1 (\w s -> w ++ ’ ’:s) ws

-- Reverse the order of all elements in a list

reverse :: [a] -> [a]

reverse = foldl (flip (:)) []

-- Compute the conjunction of a Boolean list

and :: [Bool] -> Bool

and = foldr (&&) True

-- Compute the disjunction of a Boolean list

or :: [Bool] -> Bool

or = foldr (||) False

-- Is there an element in a list satisfying a given predicate?

any :: (a -> Bool) -> [a] -> Bool

any p = or . map p

-- Is a given predicate satisfied by all elements in a list?

all :: (a -> Bool) -> [a] -> Bool

all p = and . map p

-- Element of a list?

elem :: a -> [a] -> Bool

elem x = any (x==)

-- Not element of a list?

notElem :: a -> [a] -> Bool

notElem x = all (x/=)

--- Looks up a key in an association list.

lookup :: a -> [(a,b)] -> Maybe b

lookup _ [] = Nothing

lookup k ((x,y):xys)

| k==x = Just y

| otherwise = lookup k xys

-- Generating arithmetic sequences:

enumFrom :: Int -> [Int] -- [n..]

enumFrom n = n : enumFrom (n+1)

enumFromThen :: Int -> Int -> [Int] -- [n1,n2..]

enumFromThen n1 n2 = iterate ((n2-n1)+) n1

enumFromTo :: Int -> Int -> [Int] -- [n..m]

enumFromTo n m = if n>m then [] else n : enumFromTo (n+1) m

enumFromThenTo :: Int -> Int -> Int -> [Int] -- [n1,n2..m]

enumFromThenTo n1 n2 m = takeWhile p (enumFromThen n1 n2)

where p x | n2 >= n1 = (x <= m)

| otherwise = (x >= m)

-- Conversion functions between characters and their ASCII values

ord :: Char -> Int

chr :: Int -> Char

-- Convert a term into a printable representation

show :: a -> String

-- Types of primitive arithmetic functions

62



(+) :: Int -> Int -> Int

(-) :: Int -> Int -> Int

(*) :: Int -> Int -> Int

div :: Int -> Int -> Int

mod :: Int -> Int -> Int

-- Unary minus (usually written as "- e")

negate :: Int -> Int

negate x = 0 - x

-- Constraints

-- Equational constraint

(=:=) :: a -> a -> Success

-- Always solvable constraint

success :: Success

-- Concurrent conjunction of constraints

(&) :: Success -> Success -> Success

-- Constrained expression. (c &> e) is evaluated by first solving

-- constraint c and then evaluating e.

(&>) :: Success -> a -> a

c &> x | c = x

-- Maybe type

data Maybe a = Nothing | Just a

maybe :: b -> (a -> b) -> Maybe a -> b

maybe n _ Nothing = n

maybe _ f (Just x) = f x

-- Either type

data Either a b = Left a | Right b

either :: (a -> c) -> (b -> c) -> Either a b -> c

either f _ (Left x) = f x

either _ g (Right x) = g x

-- Monadic IO

data IO a -- conceptually: World -> (a,World)

(>>=) :: IO a -> (a -> IO b) -> IO b

return :: a -> IO a

(>>) :: IO a -> IO b -> IO b

a >> b = a >>= (\_ -> b)

done :: IO ()

done = return ()

putChar :: Char -> IO ()

getChar :: IO Char

readFile :: String -> IO String

writeFile :: String -> String -> IO ()

appendFile :: String -> String -> IO ()

putStr :: String -> IO ()

putStr [] = done

putStr (c:cs) = putChar c >> putStr cs

putStrLn :: String -> IO ()

putStrLn cs = putStr cs >> putChar ’\n’

getLine :: IO String

getLine = do c <- getChar

if c==’\n’ then return []

else do cs <- getLine

return (c:cs)

-- Convert a term into a string and print it

print :: _ -> IO ()

print t = putStrLn (show t)

-- Solve a constraint as an I/O action:

-- Note: the constraint should be always solvable in a deterministic way

doSolve :: Success -> IO ()

doSolve constraint | constraint = done

-- IO monad auxiliary functions:

-- Execute a sequence of I/O actions and collect all results in a list:

sequenceIO :: [IO a] -> IO [a]

sequenceIO [] = return []

sequenceIO (c:cs) = do x <- c

xs <- sequenceIO cs

return (x:xs)

-- Execute a sequence of I/O actions and ignore the results:

sequenceIO_ :: [IO _] -> IO ()

sequenceIO_ = foldr (>>) done

-- Map an I/O action function on a list of elements.

-- The results of all I/O actions are collected in a list.

mapIO :: (a -> IO b) -> [a] -> IO [b]

mapIO f = sequenceIO . map f

-- Map an I/O action function on a list of elements.

-- The results of all I/O actions are ignored.

mapIO_ :: (a -> IO _) -> [a] -> IO ()

mapIO_ f = sequenceIO_ . map f

-- Non-determinism and free variables

-- Non-deterministic choice par excellence.

-- The value of (x ? y) is either x or y.

(?) :: a -> a -> a

x ? _ = x

_ ? y = y

-- Evaluates to a fresh free variable.

unknown :: _

unknown = let x free in x

-- Encapsulated search

-- primitive operator to control non-determinism

try :: (a->Success) -> [a->Success]

-- compute all solutions via depth-first search

solveAll :: (a->Success) -> [a->Success]

solveAll g = collect (try g)

where collect [] = []

collect [g] = [g]

collect (g1:g2:gs) = concat (map solveAll (g1:g2:gs))

-- compute first solution via depth-first search

once :: (a->Success) -> (a->Success)

once g = head (solveAll g)

-- compute all values of solutions via depth-first search

findall :: (a->Success) -> [a]

findall g = map unpack (solveAll g)

-- compute best solution via branch-and-bound with depth-first search

best :: (a->Success) -> (a->a->Bool) -> [a->Success]

-- show the solution of a solved constraint

browse :: (a->Success) -> IO ()

browse g = putStr (show (unpack g))

-- show the solutions of a list of solved constraints

browseList :: [a->Success] -> IO ()

browseList [] = done

browseList (g:gs) = browse g >> putChar ’\n’ >> browseList gs

-- unpack a solution’s value from a (solved) search goal

unpack :: (a -> Success) -> a

unpack g | g x = x where x free

C Syntax of Curry

The syntax is still preliminary and not intended to be used by automatic tools. Nevertheless, it
provides a precise definition of the concrete syntax which can be further discussed.

The syntax is close to Haskell but the following differences should be noted.

• Currently, there are no infix constructors except for “:”, the predefined list constructor. They
will be added later, although they are already used in concrete example programs.

C.1 Lexicon

The case of identifiers matters, i.e., “abc” differs from “Abc”. There are four case modes selectable
at compilation time:

• Prolog mode: variables start with an upper case latter, all other identifier symbols start with
a lower case letter.

• Gödel mode: like Prolog mode with the cases swapped.

• Haskell mode: see section 1.3 of the Haskell report.

• free: no constraints on the case of identifiers.

The default case mode is free. If a case mode is selected but not obeyed, the compiler issues a
warning.

The syntax does not define the following non-terminal symbols defining classes of identifiers:
ModuleID, TypeConstrID, DataConstrID, TypeVarID, InfixOpID, FunctionID, VariableID . All, ex-
cept InfixOpID, consist of an initial letter, whose upper or lower case depend on the case mode,
followed by any number of letters, digits, underscores, and single quotes. Additionally, Mod-
uleID can contain dots at inner positions. InfixOpID is any string of characters from the string
“~!@#$%^&*+-=<>?./|\:” or another identifier (like VariableID) enclosed in ‘. . .‘ like ‘mod‘.

The following symbols are keywords and cannot be used as an identifier:

case data do else external free if

infix infixl infixr import in let module

of then type where

Note that the symbols as, hiding, and qualified are not keywords. They have only a special
meaning in module headers and can be used as ordinary identifies.

The syntax leaves undefined Literal of primitive types. These are literal constants, such as
“1”, “3.14”, or the character “’a’”. They are as in Java which adopts the Unicode standard to
represent characters and character strings.

Comments begins either with “--” and terminate at the end of the line or with “{-” and
terminate with a matching “-}”, i.e., the delimiters “{-” and “-}” act as parentheses and can be
nested.

C.2 Layout

Similarly to Haskell, a Curry programmer can use layout information to define the structure of
blocks. For this purpose, we define the indentation of a symbol as the column number indicating
the start of this symbol. The indentation of a line is the indentation of its first symbol.13

The layout (or “off-side”) rule applies to lists of syntactic entities after the keywords let, where,
do, or of. In the subsequent context-free syntax (Section C.3), these lists are enclosed with curly
brackets ({ }) and the single entities are separated by semicolons (;). Instead of using the curly
brackets and semicolons of the context-free syntax, a Curry programmer must specify these lists
by indentation: the indentation of a list of syntactic entities after let, where, do, or of is the
indentation of the next symbol following the let, where, do, of. Any item of this list start with
the same indentation as the list. Lines with only whitespaces or an indentation greater than the
indentation of the list continue the item in its previous line. Lines with an indentation less than
the indentation of the list terminates the entire list. Moreover, a list started by let is terminated
by the keyword in. Thus, the sentence

f x = h x where {g y = y+1 ; h z = (g z) * 2 }

which is valid w.r.t. the context-free syntax, is written with the layout rules as

f x = h x

where g y = y+1

h z = (g z) * 2

or also as

f x = h x where

g y = y+1

h z = (g z)

* 2

To avoid an indentation of top-level declarations, the keyword module and the end-of-file token are
assumed to start in column 0.

C.3 Context Free Syntax

Module ::= module ModuleID [Exports] where Block
ModuleID ::= see lexicon

Exports ::= ( Export1 , . . . , Exportn )

Export ::= QFunctionName
| QTypeConstrID
| QTypeConstrID (..)

| module ModuleID
Block ::= { [ImportDecl1 ; . . . ; ImportDeclk ;]

[FixityDeclaration1 ; . . . ; FixityDeclarationm ;]
BlockDeclaration1 ; . . . ; BlockDeclarationn } (k, m, n > 0)

ImportDecl ::= import [qualified] ModuleID [as ModuleID] [ImportRestr]
13In order to determine the exact column number, we assume a fixed-width font with tab stops at each 8th column.

ImportRestr ::= ( Import1 , . . . , Importn )

| hiding ( Import1 , . . . , Importn )

Import ::= FunctionName
| TypeConstrID
| TypeConstrID (..)

BlockDeclaration ::= TypeSynonymDecl
| DataDeclaration
| FunctionDeclaration

TypeSynonymDecl ::= type SimpleType = TypeExpr
SimpleType ::= TypeConstrID TypeVarID1 . . . TypeVarIDn (n > 0)

TypeConstrID ::= see lexicon
DataDeclaration ::= data SimpleType = ConstrDecl1 | . . . | ConstrDecln (n > 0)

ConstrDecl ::= DataConstrID SimpleTypeExpr1 . . . SimpleTypeExprn (n > 0)
DataConstrID ::= see lexicon

TypeExpr ::= SimpleTypeExpr [-> TypeExpr]
SimpleTypeExpr ::= QTypeConstrID SimpleTypeExpr1 . . . SimpleTypeExprn (n > 0)

| TypeVarID | _
| () (unit type)
| ( TypeExpr1 , . . . , TypeExprn ) (tuple type, n > 1)
| [ TypeExpr ] (list type)
| ( TypeExpr ) (parenthesized type)

TypeVarID ::= see lexicon
FixityDeclaration ::= FixityKeyword Natural InfixOpID1 , . . . , InfixOpIDn (n > 0)

FixityKeyword ::= infixl | infixr | infix
Natural ::= Digit | Digit Natural

Digit ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
InfixOpID ::= see lexicon

FunctionDeclaration ::= Signature | Equat
Signature ::= FunctionNames :: TypeExpr

FunctionNames ::= FunctionName1 , . . . , FunctionNamen (n > 0)
FunctionName ::= ( InfixOpID )

| FunctionID
FunctionID ::= see lexicon

Equat ::= FunLHS = Expr [where LocalDefs]
| FunLHS CondExprs [where LocalDefs]

FunLHS ::= FunctionName SimplePat1 . . . SimplePatn (n > 0)
| SimplePat InfixOpID SimplePat

Pattern ::= QDataConstrID SimplePat1 . . . SimplePatn [: Pattern] (n > 0)
| SimplePat [: Pattern]

SimplePat ::= VariableID | _
| QDataConstrID
| Literal
| ()

| ( Pattern1 , . . . , Patternn ) (n > 1)

69



| ( Pattern )

| [ Pattern1 , . . . , Patternn ] (n > 0)
| VariableID @ SimplePat (as-pattern)

VariableID ::= see lexicon
LocalDefs ::= {ValueDeclaration1 ; . . . ; ValueDeclarationn } (n > 0)

ValueDeclaration ::= FunctionDeclaration
| PatternDeclaration
| VariableID1 , . . . , VariableIDn free (n > 0)

PatternDeclaration ::= Pattern = Expr [where LocalDefs]
CondExprs ::= | Expr = Expr [CondExprs]

Expr ::= \ SimplePat1 . . . SimplePatn -> Expr (n > 0)
| let LocalDefs in Expr (let expression)
| if Expr then Expr else Expr (conditional)
| case Expr of {Alt1 ; . . . ; Altn } (case expression, n > 0)
| do {Stmt1 ; . . . ; Stmtn ; Expr } (do expression, n > 0)
| Expr QInfixOpID Expr (infix operator)
| - Expr (unary minus)
| FunctExpr

FunctExpr ::= [FunctExpr] BasicExpr (function application)
BasicExpr ::= QVariableID (variable)

| QDataConstrID (data constructor)
| QFunctionID (defined function)
| ( QInfixOpID ) (operator function)
| Literal
| () (empty tuple)
| ( Expr ) (parenthesized expression)
| ( Expr1 , . . . , Exprn ) (tuple, n > 1)
| [ Expr1 , . . . , Exprn ] (finite list, n > 0)
| [ Expr [, Expr] .. [Expr] ] (arithmetic sequence)
| [ Expr | Qual1 , . . . , Qualn ] (list comprehension, n > 1)
| ( Expr QInfixOpID ) (left section)
| ( QInfixOpID Expr ) (right section)

Alt ::= Pattern -> Expr [where LocalDefs]
| Pattern GdAlts -> Expr [where LocalDefs]

GdAlts ::= | Expr -> Expr [GdAlts]
Qual ::= Expr

| let LocalDefs
| Pattern <- Expr

Stmt ::= Expr
| let LocalDefs
| Pattern <- Expr

QTypeConstrID ::= [ModuleID .] TypeConstrID
QDataConstrID ::= [ModuleID .] DataConstrID
QFunctionName ::= ( QInfixOpID ) | QFunctionID

QInfixOpID ::= [ModuleID .] InfixOpID
QFunctionID ::= [ModuleID .] FunctionID
QVariableID ::= [ModuleID .] VariableID

Literal ::= Int | Char | String | Float
Int ::= see lexicon

Char ::= see lexicon
String ::= see lexicon
Float ::= see lexicon

If the alternative FunctionDeclaration is used in a ValueDeclaration, then the left-hand side
(FunLHS ) must have at least one pattern after the FunctionName (instead of zero patterns which
is possible in top-level function declarations).

In CondExprs, the first expression must have type Success or Bool. In the latter case, Boolean
expressions in conditions are considered as an abbreviation for a (nested) if-then-else (see Sec-
tion 2.3.2).

In qualified names (e.g., QFunctionID), no characters (e.g., spaces) are allowed between the dot
and the module and entity names. On the other hand, an infix expression (Expr QInfixOpID Expr)
must contain at least one space or similar character after the left expression if the infix operator
starts with a dot.

C.4 Infix Operators

In the grammar above, the use of infix operators in the rule for Expr is ambiguous. These am-
biguities are resolved by assigning an associativity and precedence to each operator (InfixOpID)
by a fixity declaration. There are three kinds of associativities, non-, left- and right-associativity
(infix, infixl, infixr) and ten levels of precedence, 0 to 9, where level 0 binds least tightly
and level 9 binds most tightly. All fixity declarations must appear at the beginning of a module.
Any operator without an explicit fixity declaration is assumed to have the declaration infixl 9.
For instance, the expression “1+2*3+4==x && b” is equivalent to “((((1+(2*3))+4)==x) && b)”
w.r.t. the fixity declarations provided in the prelude.

Note that the correct use of precedences and associativities excludes some expressions which
are valid w.r.t. the context-free grammar. In general, the arguments of an infix operator must
have an infix operator at the top with a higher precedence than the current infix operator (or no
infix operator at the top). Additionaly, the left or right argument of a left- or right-associative
operator can have a top infix operator with the same precedence. The unary minus operator is
interpreted with precedence 6 which means that its argument must have a precedence of at least
7. The expression “(- t)” is not considered as a right section but as the negation of t. As a
consequence, the expressions “1<2==True” and “1 + - 3” are not allowed and must be bracketed
as “(1<2)==True” and “1 + (- 3)”.

D Operational Semantics of Curry

The precise specification of the operational semantics of Curry is based on the patterns on the rules’
left-hand sides for each function. Therefore, we describe the computation model by providing a
precise definition of pattern matching (Section D.1) which is the basis to define the computation
steps on expressions (Section D.2). The extension of this basic computation model to solving equa-
tional constraints and higher-order functions is described in Sections D.3 and D.4, respectively.
Section D.5 describes the automatic generation of the definitional trees to guide the pattern match-
ing strategy. Finally, Section D.6 specifies the operational semantics of the primitive operator try
for encapsulating search.

D.1 Definitional Trees

We start by considering only the unconditional first-order part of Curry, i.e., rules do not contain
conditions and λ-abstractions and partial function applications are not allowed. We assume some
familiarity with basic notions of term rewriting [15] and functional logic programming [22].

We denote by C the set of constructors (with elements c, d), by F the set of defined functions
or operations (with elements f, g), and by X the set of variables (with elements x, y), where C, F
and X are disjoint. An expression (data term) is a variable x ∈ X or an application ϕ(e1, . . . , en)
where ϕ ∈ C ∪ F (ϕ ∈ C) has arity n and e1, . . . , en are expressions (data terms).15 The set of all
expressions and data terms are denoted by T (C ∪ F ,X ) and T (C,X ), respectively. A call pattern
is an expression of the form f(t1, . . . , tn) where each variable occurs only once, f ∈ F is an n-ary
function, and t1, . . . , tn ∈ T (C,X ). root(e) denotes the symbol at the root of the expression e.
A position p in an expression e is represented by a sequence of natural numbers, e|p denotes the
subterm or subexpression of e at position p, and e[e′]p denotes the result of replacing the subterm
e|p by the expression e′ (see [15] for details).

A substitution is a mapping σ:X → T (C ∪F ,X ) with σ(x) 6= x only for finitely many variables
x. Thus, we denote a substitution by σ = {x1 7→ t1, . . . , xn 7→ tn}, where σ(xi) 6= xi for i = 1, . . . , n,
and id denotes the identity substitution (n = 0). Dom(σ) = {x1, . . . , xn} is the domain of σ and

VRan(σ) = {x | x is a variable occurring in some ti, i ∈ {1, . . . , n}}

is its variable range. Substitutions are extended to morphisms on expressions by σ(f(e1, . . . , en)) =
f(σ(e1), . . . , σ(en)) for every expression f(e1, . . . , en).

A Curry program is a set of rules l = r satisfying some restrictions (see Section 2.3). In particular,
the left-hand side l must be a call pattern. A rewrite rule is called a variant of another rule if it is
obtained by a unique replacement of variables by other variables.

An answer expression is a pair σ e consisting of a substitution σ (the answer computed so far)
and an expression e. An answer expression σ e is solved if e is a data term. We sometimes omit
the identity substitution in answer expressions, i.e., we write e instead of id e if it is clear from the
context. A disjunctive expression is a (multi-)set of answer expressions {σ1 e1, . . . , σn en}. The
set of all disjunctive expressions is denoted by D.

15Since we do not consider partial applications in this part, we write the full application as ϕ(e1, . . . , en) which is

equivalent to Curry’s notation ϕ e1 . . . en.

A single computation step performs a reduction in exactly one unsolved expression of a dis-
junction. To provide a precise definition of the operational semantics, we use definitional trees.16

A definitional tree is a hierarchical structure containing all rules of a defined function. T is a
definitional tree with call pattern π iff the depth of T is finite and one of the following cases holds:

T = rule(l = r), where l = r is a variant of a rule such that l = π.

T = branch(π, o, r, T1, . . . , Tk), where o is an occurrence of a variable in π, r ∈ {rigid, flex},
c1, . . . , ck are different constructors of the sort of π|o, for some k > 0, and, for all i = 1, . . . , k,
Ti is a definitional tree with call pattern π[ci(x1, . . . , xn)]o, where n is the arity of ci and
x1, . . . , xn are new variables.

T = or(T1, T2), where T1 and T2 are definitional trees with call pattern π.17

The rigid/flex tag in a branch will be used to specify the suspension behavior w.r.t. free variables
as actual arguments. User-defined functions are always translated into definitional trees with flex

tags, but case expressions are translated into definitional trees with rigid tags.
A definitional tree of an n-ary function f is a definitional tree T with call pattern f(x1, . . . , xn),

where x1, . . . , xn are distinct variables, such that for each rule l = r with l = f(t1, . . . , tn) there is a
node rule(l′ = r′) in T with l variant of l′. In the following, we write pat(T ) for the call pattern of
a definitional tree T .

It is always possible to construct a definitional tree for each function (concrete algorithms are
described in [4, 33] and in Section D.5). For instance, consider the following definition of the
less-or-equal predicate on natural numbers represented by data terms built from z (zero) and s

(successor):

data Nat = z | s Nat

leq :: Nat -> Nat -> Bool

leq z n = True

leq (s m) z = False

leq (s m) (s n) = leq m n

Consider a function call like (leq e1 e2). In order to apply some reduction rule, the first argument
e1 must always be evaluated to head normal form (i.e., to an expression without a defined function
symbol at the top). However, the second argument must be evaluated only if the first argument
has the form (s e).18 This dependency between the first and the second argument is expressed by
the definitional tree

branch(leq(x1, x2), 1, f lex, rule(leq(z, x2) = True),
branch(leq(s(x), x2), 2, f lex, rule(leq(s(x), z) = False),

rule(leq(s(x), s(y)) = leq(x, y))))

16Our notion is influenced by Antoy’s work [4], but here we use an extended form of definitional trees.
17For the sake of simplicity, we consider only binary or nodes. The extension to such nodes with more than two

subtrees is straightforward.
18Naive lazy narrowing strategies may also evaluate the second argument in any case. However, as shown in [7],

the consideration of dependencies between arguments is essential to obtain optimal evaluation strategies.

Computation step for a single expression:

Eval[[ei]] ⇒ D

Eval[[e1 & e2]] ⇒ replace(e1 & e2, i,D)
i ∈ {1, 2}

Eval[[ei]] ⇒ D

Eval[[c(e1, . . . , en)]] ⇒ replace(c(e1, . . . , en), i,D)
i ∈ {1, . . . , n}

Eval[[f(e1, . . . , en); T ]] ⇒ D

Eval[[f(e1, . . . , en)]] ⇒ D
if T is a definitional tree for f with fresh variables

Computation step for an operation-rooted expression e:

Eval[[e; rule(l = r)]] ⇒ {id σ(r)}
if σ is a substitution with σ(l) = e

Eval[[e; T1]] ⇒ D1 Eval[[e; T2]] ⇒ D2

Eval[[e; or(T1, T2)]] ⇒ D1 ∪D2

Eval[[e; branch(π, p, r, T1, . . . , Tk)]] ⇒
D if e|p = c(e1, . . . , en), pat(Ti)|p = c(x1, . . . , xn), and Eval[[e; Ti]] ⇒ D

∅ if e|p = c(· · ·) and pat(Ti)|p 6= c(· · ·), i = 1, . . . , k⋃k
i=1{σi σi(e)} if e|p = x, r = flex, and σi = {x 7→ pat(Ti)|p}

replace(e, p, D) if e|p = f(e1, . . . , en) and Eval[[e|p]] ⇒ D

Derivation step for a disjunctive expression:
Eval[[e]] ⇒ {σ1 e1, . . . , σn en}

{σ e} ∪D ⇒ {σ1 ◦ σ e1, . . . , σn ◦ σ en} ∪D

Figure 2: Operational semantics of Curry

This definitional tree specifies that the first argument is initially evaluated and the second argument
is only evaluated if the first argument has the constructor s at the top. The precise operational
meaning induced by definitional trees is described in the following section.

D.2 Computation Steps

The operational semantics is defined by the derivability relation D1 ⇒ D2 on disjunctive expressions

specified in Figure 2. An inference rule
α

β
should be read as “if α is derivable, then β is also

derivable”. We say that the computation of an expression e suspends if there is no D with Eval[[e]] ⇒
D. A constraint is solvable if it can be reduced to success. The exact method to solve constraints
depends on the constraint solver. A method to solve equational constraints on data terms is
specified in Section D.3. For the purpose of this definition, we consider success as the neutral
element of the operation &, i.e., success & c and c & success are considered as equivalent to c.

As can be seen by the rules for evaluating constraints in Figure 2, the concurrent conjunction
of two constraints is evaluated by applying computation steps to one of the two constraints. Since
both constraints must be finally solved (i.e., reduced to success), one can consider the evalua-

tion of a concurrent conjunction as two computation threads evaluating these constraints. They
are synchronized by accessing common variables (which may suspend a thread, see below). In a
simple sequential implementation, the evaluation of e1 & e2 could be started by an attempt to solve
e1. If the evaluation of e1 suspends, an evaluation step is applied to e2. If a variable responsible
to the suspension of e1 is instantiated, the left expression e1 will be evaluated in the subsequent
step. Thus, we obtain a concurrent behavior with an interleaving semantics. However, a sophis-
ticated implementation should provide a fair selection of threads, e.g., as done in a Java-based
implementation of Curry [26].

A computation step for an expression e attempts to apply a reduction step to an outermost
operation-rooted subterm in e by evaluating this subterm with the definitional tree for the function
symbol at the root of this subterm. Although it is unspecified which outermost subterm is evaluated
(compare second inference rule in Figure 2 for constructor-rooted expressions), we assume that a
single (e.g., leftmost) outermost subterm is always selected. The evaluation of an operation-rooted
term is defined by a case distinction on the definitional tree. If it is a rule node, this rule is applied.
An or node produces a disjunction where the different alternatives are combined. Note that this
definition requires that the entire disjunction suspends if one disjunct suspends. This can be relaxed
in concrete implementations by continuing the evaluation in one or branch even if the other branch
is suspended. However, to ensure completeness, it is not allowed to omit a suspended or branch
and continue with the other non-suspended or branch [24]. For a similar reason, we cannot commit
to a disjunct which does not bind variables but we have to consider both alternatives (see [6] for a
counter-example).

The most interesting case is a branch node. Here we have to branch on the value of the
top-level symbol at the selected position. If the symbol is a constructor, we proceed with the
appropriate definitional subtree, if possible. If it is a function symbol, we proceed by evaluating
this subexpression. If it is a variable and the branch is flexible, we instantiate the variable to
the different constructors, otherwise (if the branch is rigid) we cannot proceed and suspend. The
auxiliary function replace puts a possibly disjunctive expression into a subterm:

replace(e, p, {σ1 e1, . . . , σn en}) = {σ1 σ1(e)[e1]p, . . . , σn σn(e)[en]p} (n ≥ 0)

The overall computation strategy on disjunctive expressions takes a disjunct σ e not in solved
form and evaluates e. If the evaluation does not suspend and yields a disjunctive expression, we
substitute it for σ e composed with the old answer substitution.

Soundness and completeness results for the operational semantics can be found in [24]. Note
that the evaluation of expressions is completely deterministic if the concurrent conjunction of
constraints does not occur.

Conditional rules. Note that the operational semantics specified in Figure 2 handles only un-
conditional rules. Conditional rules are treated by the rules in Figure 3. If a conditional rules
is applied, the condition and the right-hand side are joined into a guarded expression, i.e., a pair
of the form c|r. Due to the additional evaluation rules, a guarded expression is evaluated by an
attempt to solve its condition. If the condition is solvable (i.e., reducible to success), the guarded
expression is replaced by its right-hand side. If the condition is not solvable and does not suspend,
the disjunct containing this guarded expression will be deleted by the definition of computation
steps, i.e., the application of the conditional rule fails and does not contribute to the final result.

Eval[[e; rule(l | c = r)]] ⇒ {id σ(c|r)}
if σ is a substitution with σ(l) = e

Eval[[c]] ⇒ D

Eval[[c|e]] ⇒ replace(c|e, 1, D) Eval[[success|e]] ⇒ {id e}

Figure 3: Evaluation of conditional rules

Eval[[c(e1, . . . , en)]] ⇒ D

Eval[[ensureNotFree(c(e1, . . . , en))]] ⇒ D

Eval[[f(e1, . . . , en)]] ⇒ D

Eval[[ensureNotFree(f(e1, . . . , en))]] ⇒ replace(ensureNotFree(f(e1, . . . , en)), 1, D)

Figure 4: Evaluation of ensureNotFree

Ensuring instantiation The semantics of the primitive ensureNotFree to ensure computations
with instantiated expressions, as informally described in Section 5.4, can be easily described by the
rules in Figure 4. If ensureNotFree occurs in a computation with a constructor term as argument,
it has no effect on the computation (first rule). It behaves like the identity as long as the argument
is a function call. Otherwise (i.e., if the argument is a free variable), no evaluation step is possible.

Sharing and graph rewriting. For the sake of simplicity, this description of the operational
semantics is based on term rewriting and does not take into account that common subterms are
shared (see Section 2.3.1). We only note here that several occurrences of the same variable are
always shared, i.e., if an argument of a function is instantiated during a call to this function to
an expression and this expression is evaluated to some value (head normal form), then all other
expressions resulting from instantiating occurrences of the same argument are replaced by the
same value (head normal form).19 This is necessary not only for efficiency reasons but also for the
soundness of the operational semantics in the presence of non-deterministic functions, as discussed
in Section 2.3.1. The sharing of variables can be described with the more complicated framework
of graph rewriting. Formal descriptions of graph rewriting and narrowing can be found in [16, 19].
A formal description of Curry’s operational semantics based on heap structures to model sharing
can be found in [3].

D.3 Equational Constraints

An equational constraint e1=:=e2 is solvable if both sides are reducible to a unifiable data term
(strict equality). An equational constraint can be solved in an incremental way by an interleaved

19 It should be noted that values are constructor terms like 23, True, or [2,4,5]. This means that the eval-

uation of constraints and I/O actions are not shared since they are not replaced by a value after evaluation but

constraints are solved in order to apply a conditional rule (in case of constraints) and I/O actions are applied

to the outside world when they appear at the top-level in a program. This is the intended behavior since the

expressions “putChar ’a’ >> putChar ’a’” and “let ca = putChar ’a’ in ca >> ca” should have an identical

behvarior, namely printing the character ’a’ twice.

Eval[[ei]] ⇒ D

Eval[[e1=:=e2]] ⇒ replace(e1=:=e2, i,D)
if ei = f(t1, . . . , tn), i ∈ {1, 2}

Eval[[c(e1, . . . , en)=:=c(e′1, . . . , e
′
n)]] ⇒ {id e1=:=e′1 & . . . & en=:=e′n}

Eval[[c(e1, . . . , en)=:=d(e′1, . . . , e
′
m)]] ⇒ ∅

if c 6= d or n 6= m

Eval[[x=:=e]] ⇒ D

Eval[[e=:=x]] ⇒ D
if e is not a variable

Eval[[x=:=y]] ⇒ {{x 7→ y} success}

Eval[[x=:=c(e1, . . . , en)]] ⇒ {σ y1=:=σ(e1) & . . . & yn=:=σ(en)}

if x 6∈ cv(c(e1, . . . , en)),
σ = {x 7→ c(y1, . . . , yn)},
y1, . . . , yn fresh variables

Eval[[x=:=c(e1, . . . , en)]] ⇒ ∅
if x ∈ cv(c(e1, . . . , en))

Figure 5: Solving equational constraints

lazy evaluation of the expressions and binding of variables to constructor terms. The evaluation
steps implementing this method are shown in Figure 5, where we consider the symbol =:= is a
binary infix function symbol. The last two rules implements an occur check where the critical
variables (cv) are only those variables occurring outside a function call:

cv(x) = {x}
cv(c(e1, . . . , en)) = cv(e1) ∪ . . . ∪ cv(en)

cv(f(e1, . . . , en)) = ∅

However, Curry implementations can also provide other methods to solve equational constraints
or other types of constraints with appropriate constraint solvers. The only property, which is
important for the operational semantics, is the fact that a solved constraint has the form success,
i.e., a solvable constraint c is reducible to the answer expression σ success where σ is a solution
of c.

D.4 Higher-Order Features

Warren [49] has shown for the case of logic programming that the higher-order features of functional
programming can be implemented by providing a (first-order) definition of the application function.
Since Curry supports the higher-order features of current functional languages but excludes the
guessing of higher-order objects by higher-order unification (as, e.g., in [25, 38, 43]), the operational
semantics specified in Figure 2 can be simply extended to cover Curry’s higher-order features by
adding the following axiom (here we denote by the infix operator “@” the application of a function

77



to an expression):

Eval[[ϕ(e1, . . . , em)@e]] ⇒ ϕ(e1, . . . , em, e) if ϕ has arity n and m < n

Thus, we evaluate an application by simply adding the argument to a partial application. If a
function has the right number of arguments, it is evaluated by the rules in Figure 2. Note that
the function application suspends if the applied function is unknown (instead of a possible but
expensive non-deterministic search for the appropriate function).

λ-abstractions occurring in expressions are anonymous functions and can be implemented by
assigning a new name to each λ-abstraction, where the free variables become parameters of the
function. For instance, the λ-abstraction (f is a free variable)

\x y -> f x y + 2

can be replaced by the (partial) application (lambda f), where lambda is a new name, and adding
the rule

lambda f x y = f x y + 2

Note that this transformation (as well as the extension of Eval[[·]] above) is only used to explain the
operational meaning of Curry’s higher-order features. A concrete Curry system is free to choose
other implementation techniques.

D.5 Generating Definitional Trees

Curry’s computation model is based on the specification of a definitional tree for each operation.
Although definitional trees are a high-level formalism to specify evaluation strategies, it is tedious
to annotate each operation which its definitional tree. Therefore, the user need not write them
explicitly in the program since they are automatically inserted by the Curry system. In the fol-
lowing, we present Curry’s algorithm to generate definitional trees from the left-hand sides of the
functions’ rules.

The generation of definitional trees for each function is not straightforward, since there may
exist many non-isomorphic definitional trees for a single function representing different evaluation
strategies. This demands for a default strategy to generate definitional trees. Curry uses the
following default strategy:

1. The leftmost argument, where a constructor occurs at the corresponding position in all left-
hand sides defining this function, is evaluated to head normal form.

2. or nodes (i.e., disjunctions) are generated in case of a conflict between constructors and
variables in the left-hand sides, i.e., if two rules have a variable and a constructor at the same
position on the left-hand side.

In the following, we assume that all rules are unconditional (it is obvious how to extend it to
conditional rules since only the left-hand sides of the rules are relevant for the definitional trees).
To specify the construction algorithm, we define by

DP (π,R) = {o position of a variable in π | root(l|o) ∈ C for some l = r ∈ R}

the set of demanded positions of a call pattern π w.r.t. a set of rules R. For instance, the demanded
positions of the call pattern leq(x,y) w.r.t. the rules for the predicate leq (see Section 15, page 73)
are {1, 2} referring to the pattern variables x and y. Furthermore, we define by

IP (π,R) = {o ∈ DP (π,R) | root(l|o) ∈ C for all l = r ∈ R}

the set of inductive positions of a call pattern π w.r.t. a set of rules R. Thus, the inductive
positions are those demanded positions where a constructor occurs in all left-hand sides defining
this function. For instance, the set of inductive positions of the call pattern leq(x,y) w.r.t. the
rules for the predicate leq is {1}.

The generation of a definitional tree for a call pattern π and a set of rules R (where l is an
instance of π for each l = r ∈ R) is described by the function gt(π,R). We distinguish the following
cases for gt:

1. If the position o is leftmost in IP (π,R), {root(l|o) | l = r ∈ R} = {c1, . . . , ck} where c1, . . . , ck

are different constructors with arities n1, . . . , nk, and Ri = {l = r ∈ R | root(l|o) = ci}, then

gt(π,R) = branch(π, o, flex, gt(π[c1(x11, . . . , x1n1)]o, R1),
. . . ,

gt(π[ck(xk1, . . . , xknk
)]o, Rk))

where xij are fresh variables. I.e., we generate a branch node for the leftmost inductive
position (provided that there exists such a position).

2. If IP (π,R) = ∅, let o be the leftmost position in DP (π,R) and R′ = {l = r ∈ R | root(l|o) ∈
C}. Then

gt(π,R) = or(gt(π,R′), gt(π,R−R′))

I.e., we generate an or node if the leftmost demanded position of the call pattern is not
demanded by the left-hand sides of all rules.

3. If DP (π,R) = ∅ and l = r variant of some rule in R with l = π, then

gt(π,R) = rule(l = r)

Note that all rules in R are variants of each other if there is no demanded position (this
follows from the weak orthogonality of the rewrite system). For non-weakly orthogonal rewrite
systems, which may occur in the presence of non-deterministic functions [19] or conditional
rules, the rules in R may not be variants. In this case we simply connect the different rules
by or nodes.

If R is the set of all rules defining the n-ary function f , then a definitional tree for f is generated
by computing gt(f(x1, . . . , xn), R).

It is easy to see that this algorithm computes a definitional tree for each function since the
number of rules is reduced in each recursive call and it keeps the invariant that the left-hand sides
of the current set of rules are always instances of the current call pattern.

Eval[[try(g)]] ⇒

[] if Eval[[c]] ⇒ ∅

[g′] if Eval[[c]] ⇒ {σ success} (i.e., σ is a mgu for all equations in c) with
Dom(σ) ⊆ {x, x1, . . . , xn} (or c = success and σ = id) and
g′ = \x->let x1, . . . , xn free in σ ◦ ϕ success

D if Eval[[c]] ⇒ {σ c′} with Dom(σ) ⊆ {x, x1, . . . , xn},
g′ = \x->let x1, . . . , xn, y1, . . . , ym free in σ ◦ ϕ c′

where {y1, . . . , ym} = VRan(σ) \ ({x, x1, . . . , xn} ∪ free(g)),
and Eval[[try(g′)]] ⇒ D

[g1,...,gk] if Eval[[c]] ⇒ {σ1 c1, . . . , σk ck}, k > 1, and, for i = 1, . . . , k,
Dom(σi) ⊆ {x, x1, . . . , xn} and
gi = \x->let x1, . . . , xn, y1, . . . , ymi free in σi ◦ ϕ ci

where {y1, . . . , ymi} = VRan(σi) \ ({x, x1, . . . , xn} ∪ free(g))

Figure 6: Operational semantics of the try operator for g = \x->let x1, . . . , xn free in ϕ c

Note that the algorithm gt is not strictly conform with the pattern matching strategy of func-
tional languages like Haskell. For instance, it generates for the rules

f 0 0 = 0

f x 1 = 0

the definitional tree

branch(f(x1,x2), 2, f lex, branch(f(x1,0), 1, f lex, rule(f(0,0) = 0)),
rule(f(x1,1) = 0))

(although both arguments are demanded, only the second argument is at an inductive position)
whereas Haskell has a strict left-to-right pattern matching strategy which could be expressed by
the definitionl tree

or(branch(f(x1,x2), 1, f lex, branch(f(0,x2), 2, f lex, rule(f(0,0) = 0))),
branch(f(x1,x2), 2, f lex, rule(f(x1,1) = 0)))

The latter tree is not optimal since it has a non-deterministic or node and always requires the
evaluation of the first argument (in the first alternative). If the function definitions are uniform in
the sense of [48], the strategy described by gt is identical to traditional functional languages.

D.6 Encapsulated Search

The exact behavior of the try operator is specified in Figure 6. Note that a substitution ϕ of the
form {x1 7→ t1, . . . , xn 7→ tn}, computed by evaluating the body of a search goal, must be encoded as
a constraint in the new search goal. Therefore, ϕ is transformed into its equational representation
ϕ, where ϕ is a solved constraint of the form x1=:=t1 &...& xn=:=tn. Hence, a search goal can
always be written as try (\x->ϕ & c). Since the solved constraint ϕ does not have to be evaluated
again, we use the sugared form try (\x->ϕ c) where only c is further evaluated. Thus, an initial
search goal try (\x->c) is equivalent to try (\x->id c).

A search goal is solved (second case) if the constraint is solvable without bindings of global
variables. In a deterministic step (third case), we apply the try operator again after adding the
newly introduced variables to the list of local variables. Note that the free variables occurring in
g, denoted by free(g), must not be locally declared because they can appear also outside of g, and
therefore they have to be removed from VRan(σ). In a non-deterministic step (fourth case), we
return the different alternatives as the result. Note that the evaluation of the try operator suspends
if a computation step on the constraint attempts to bind a free variable. In order to ensure that an
encapsulated search will not be suspended due to necessary bindings of free variables, the search
goal should be a closed expression when a search operator is applied to it, i.e., the search variable
is bound by the lambda abstraction and all other free variables are existentially quantified by local
declarations.

If one wants to use free variables in a search goal, e.g., for synchronization in systems where
concurrency is intensively used, the goal should perform only deterministic steps until all free
variables have been bound by another part of the computation. Otherwise, a non-deterministic
step could unnecessarily split the search goal. For instance, consider the expression

try (\x -> y=:=[x] & append x [0,1] =:= [0,1])

The computation of y=:=[x] suspends because y is free. Therefore, a concurrent computation of
the conjunction will split the goal by reducing the append expression. However, if y is first bound
to [[]] by another thread of the computation, the search goal could proceed deterministically.

To avoid such unnecessary splittings, we can restrict the applicability of the split case and adapt
a solution known as stability from AKL [28] and Oz [45]. For this purpose, we can slightly change
the definition of try such that non-deterministic steps lead to a suspension as long as a deterministic
step might be enabled by another part of the computation, i.e., the search goal contains free global
variables. To do so, we have to replace only the fourth rule of the try operator:

Eval[[try(g)]] ⇒ [g1,...,gk] if Eval[[c]] ⇒ {σ1 c1, . . . , σk ck}, k > 1,

Eval[[c]] 6⇒ {σ c′}, free(c) ⊆ {x, x1, . . . , xn}
and, for i = 1, . . . , k,

gi = \x->let x1, . . . , xn, y1, . . . , ymi free in σi ◦ ϕ ci

where y1, . . . , ymi = VRan(σi) \ ({x, x1, . . . , xn} ∪ free(g))

Thus, a non-deteterministic step is only performed if no deterministic step is possible and the
unsolved constraint c contains no free variables except those locally declared in the search goal.
Global variables appearing in the data terms ti of the substitution ϕ = {x1 7→ t1, . . . , xn 7→ tn} but
not in c are not considered by these conditions because they cannot influence the further evaluation
of c. Note that the former condition Dom(σi) ⊆ {x, x1, . . . , xn} is covered by the new condition
free(c) ⊆ {x, x1, . . . , xn} because Dom(σi) is always a subset of free(c).

Note that the first definition of try in Figure 6 computes non-deterministic splittings where
the latter definition suspends. On the other hand, the latter definition can avoid some unnecessary
splittings. Thus, different implementations of Curry can support one or both of these definitions.

D.7 Eliminating Local Declarations

In the description of the operational semantics above, we assumed that a Curry program is a set
of rules l = r where the left-hand side l is a call pattern and the right-hand side r is an expression

containing variables, constructors, and defined functions. However, a Curry program can also
contain local declarations (cf. Section 2.4) whose operational semantics is not defined so far. To
simplify the operational semantics, we do not extend it to local declarations but provide in the
following a transformation of Curry programs containing local declarations into Curry programs
without local declarations (except for free variables). The main purpose of this transformation is to
provide a precise definition of the operational semantics of the full language. This transformation
can also be performed by implementations of Curry but it is also possible that some implementations
provide explicit support for local declarations, provided that they satisfy the operational meaning
described in the following.

The elimination of local function and pattern declarations is done by the following steps which
are subsequently described in more detail:

1. Eliminate sections and lambda abstractions

2. Eliminate Boolean guards in rules

3. Transform where declarations into let declarations

4. Eliminate local patterns and functions

For the following, we assume that all name conflicts have been resolved, i.e., the names of functions
and argument variables defined in the different declarations are pairwise different.

Eliminate sections and lambda abstractions:
Each left section (expr op) is transformed into the partial application ((op) expr).

Each right section (op expr) is transformed into the lambda abstraction (\x->(op) x expr)

where x is a new variable.

All λ-abstractions are eliminated by providing a new name (cf. Section D.4), i.e., each λ-
abstraction (\x1 . . . xn->e) is replaced by the expression (let f x1 . . . xn = e in f) where
f is a new function name. This transformation must be recursively applied to all λ-
abstractions occurring in e.

Eliminate Boolean guards:
This is done according the meaning described in Section 2.3.2. A rule of the form

f t1 . . . tn | b1 = e1
...

| bk = ek

where decls

(the where part can also be missing), where all guards b1, . . . , bk (k > 0) are expressions of
type Bool, is transformed into the single rule

f t1 . . . tn = if b1 then e1 else
...

if bk then ek else failed

where decls

Transform where into let:
Each unconditional rule of the form

l = r where decls

is transformed into

l = let decls in r .

Each conditional rule of the form

l | c = r [ where decls ]

is transformed into

l = [ let decls in ] (c|r)

where the meaning of the guarded expression (c|r) is explained in Figure 3. Thus, we assume
in the subsequent transformations that all program rules are of the form “l = r” (where r might
be a let-expression).

Note that this transformation is not really necessary but provides for a unified treatment of
the elimination of local pattern and function declarations in let and where.

Eliminate local patterns and functions:
All local pattern and function declarations in a rule “l = r” are eliminated by iterating the
elimination of outermost local declarations as follows. For this purpose, we denote by r =
C[let decls in e]p an outermost let declaration in r, i.e., there is no prefix position p′ < p

in r with r|p′ = let . . .

We apply the following transformations as long as possible to eliminate all local pattern and
function declarations in a rule of the form “l = C[let decls in e]p”:

Eliminate patterns:
Select a local pattern declaration which contains only argument variables from the main
function’s left-hand side in the expression, i.e., the rule has the form

l = C[let decls1

p = e′

decls2

in e]p

with free(e′) ⊆ free(l) (it is a programming error if no pattern declaration has this
property, i.e., cyclic pattern definitions or pattern definitions depending on locally free
variables are not allowed). Then transform this rule into the rules

l = C[f ′ x1 . . . xk e′]p
f ′ x1 . . . xk z = f ′′ x1 . . . xk (f1 z) . . . (fm z)
f ′′ x1 . . . xk y1 . . . ym = let decls1

decls2

in e

f1 p = y1

...

fm p = ym

where x1 . . . xk are all the variables occurring in l, y1 . . . ym are all the variables occurring
in p, z is a new variable symbol, and f ′, f ′′, f1, . . . , fm are new function symbols. Repeat
this step for f ′′ until all local pattern declarations are eliminated.

This translation can be optimized in some specific cases. If p is just a variable, the
function f ′ is not needed and the definition of l can be simplified into

l = C[f ′′ x1 . . . xk e′]p

Similarly, if e′ is a variable, the function f ′ is also not needed and the definition of l can
be replaced by

l = C[f ′′ x1 . . . xk (f1 e′) . . . (fm e′)]p

Complete local function definitions:
If a locally declared function f refers in its definition to a variable v not contained in
its argument patterns, then add v as an additional first20 argument to all occurrences
of f (i.e., left-hand sides and calls). Repeat this step until all locally defined functions
are completed. Note that this completion step must also be applied to free variables
introduced in the same let expression, i.e., the rule

f x = let y free

g z = c y z

in g 0

is completed to

f x = let y free

g y z = c y z

in g y 0

Globalize local function definitions:
If the definitions of all locally declared functions are completed, i.e., the definitions only
refer to variables in the argument patterns, delete the locally declared functions and
define them at the top level (and rename them if there are already top-level functions
with the same name). For instance, the previous rule is transformed into the definitions

g y z = c y z

f x = let y free in g y 0

Note that the entire transformation process must be applied again to the new top-level
declarations since they may also contain local declarations.

After applying these transformation steps to all rules in the program, we obtain a program without
sections and λ-abstractions where all local declarations contain only free variables. Note that
partial applications are not eliminated since they can be treated as shown in Section D.4.

20The new arguments must come first because of possible partial applications of this function.

The final program might not be a valid Curry program due to the transformation of conditional
rules. In order to obtain a valid Curry program, we can perform the following final transformation:

Transform l = let decls in (c|r) into l | c = r where decls

